Taking Care of Business with Your COMMODORE 64®

David P. Dautenhahn

More than 100 brief, easy-to-use BASIC programs to help you get the most out of your money

650.028
D26T

HAYDEN

Digitized by the Internet Archive
in 2012

http://www.archive.org/details/takingcareofbusi00daut

STO

650.028 D26t 7073591
Dautenhahn, David P.
Taking care of business with
your Commodore 64

DO NOT REMOVE
CARDS FROM POCKET

ALLEN COUNTY PUBLIC LIBRARY
FORT WAYNE, INDIANA 46802

You may return this book to any agency, branch,
or bookmobile of the Allen County Public Library.

DEMCO

Taking Care of Business with Your Commodore 64®

David P. Dautenhahn

HAYDEN BOOK COMPANY
a division of Hayden Publishing Company, Inc.
Hasbrouck Heights, New Jersey

Allen County Public Library
Ft. Wayne, Indiana

Acquisitions Editor: M.C. VARLEY
Developmental Editor: KAREN PASTUZYN
Production Editor: LORI WILLIAMS
Designer: JIM BERNARD
Cover Illustration: PETER BONO
Compositor: AMERICAN-STRATFORD GRAPHIC SERVICES, INC.
Printed and Bound by: ARCATA GRAPHICS CO.:
FAIRFIELD GRAPHICS DIVISION

Library of Congress Cataloging in Publication Data

Dautenhahn, David P.
Taking care of business with your Commodore 64.

1. Commodore 64 (Computer)—Programming. 2. Basic (Computer program language) 3. Business—Data processing.
I. Title.
HF5548.4.C65D38 1984 650'.028'54 84-22546
ISBN 0-8104-6457-8

Commodore 64 is a registered trademark of Commodore Business Machines, Inc., which is not affiliated with Hayden Book Company.

Copyright © 1985 by HAYDEN BOOK COMPANY. All rights reserved. No part of this book may be reprinted, or reproduced, or utilized in any form or by any electronic, mechanical, or other means, now known or hereafter invented, including photocopying and recording, or in any information storage and retrieval system, without permission in writing from the Publisher.

Printed in the United States of America

1	2	3	4	5	6	7	8	9	
85	86	87	88	89	90	91	92	93	YEAR

Acknowledgments

7073591

I would like to thank Chris Varley of Hayden Book Company for his assistance in the development of this book and my dear wife Karla for her help with typing, spelling, ideas, and encouragement. I would also like to thank Sarah and Jeremy for allowing me the time to write this book.

Preface

This book is designed for the personal computer user who is interested in implementing business and financial applications for needs at home, school, or business. It contains many of the standard financial formulas and equations presented in a programmed structure.

This book is also valuable as a reference source for equations that may be necessary when writing other business-related programs, and it may help initiate ideas for programs that could assist you in your business or personal needs.

The material in this book is designed for both the novice and the advanced user. It has many facets for both levels of users and could serve as an extremely valuable resource in almost any computer library.

The Commodore 64 is a very flexible and powerful microcomputer. It has 64K of RAM, as the name implies, and quite a wide range of peripherals. The programs presented in this book require just the base system and no additional system peripheral devices.

Many of the programs in this book can be combined into a customized application for your specific business or financial needs. Appendix A explains how this can be done and shows an example of how numerous functions can be called from a general menu of available options. It should also be noted that the programs shown in this book can be run on any computer that uses the BASIC language. This may require a slight modification to the syntax of the program.

Each topic in the book is presented in four parts as follows:

1. The application is presented with a short explanation of the subject.
2. The BASIC program is then listed.
3. A real-life scenario for the application is given.
4. A sample run of the program with the previous scenario is then shown.

There are always some differences of opinion in the interpretation of some financial applications. The author and publisher do not guarantee any specific results from the programs herein.

David P. Dautenhahn

Contents

Chapter 4. Real Estate 28

Chapter 5. Loan Analysis 50

Chapter 6. Savings 61

Taking Care of Business with Your Commodore 64®

CHAPTER

1

Interest

When borrowing or lending money, a very important factor of the transaction is the interest involved. You can invest your money in a growth situation so that it earns interest, or you can borrow money from a lending institution, to which you must pay interest for using its money. Analysis of interest being paid or received is a common occurrence in almost everyone's daily activities. Current interest rates must be monitored closely along with their effect upon your investments and loans.

The amount of interest paid depends on four factors:

1. the amount of money involved, usually called the principal;
2. the interest rate, usually expressed as an annual rate;
3. the length of time of the loan;
4. the method of interest calculation, several of which are shown in this chapter.

The standard compound interest equation is shown below:

$$FV = PV \times \left(1 + \frac{I}{C}\right)^{(C \times T)}$$

where: FV = *future value*
PV = *present value*
I = *interest rate*
C = *number of compound periods per year*
T = *number of years compounded*

When there is only one compound period per year the equation is denoted as follows:

$$FV = PV \times (1 + I)^{T}$$

The programs presented in this chapter solve this standard compound interest equation for numerous variables.

Interest is also an important aspect of a savings or loan situation where periodic payments are made. The future value of an account where periodic payments are made is shown with the following equation:

$$FV = PM \times \left[\frac{(1 + I)^{N} - 1}{I} \right]$$

where: FV = *future value*
PM = *periodic payment*
I = *periodic interest rate*
N = *number of payment periods*

The interest rate with this equation is a periodic interest rate. It represents an interest rate for a specific period of time. If monthly payments are made, the periodic interest rate would be the annual rate divided by 12.

It should be noted that the interest rate in all programs in this chapter should be entered as a percentage value and should not be entered in its decimal format.

SIMPLE INTEREST

This program will prompt for the required variables and compute the simple interest. The formula for computing simple interest is the principal multiplied by the interest rate and time period. Simple interest is different from compound interest in that the latter earns you interest on the interest previously earned. With simple interest you earn interest only on the initial deposit. The following is a simple interest program.

```
10 INPUT"ENTER AMOUNT";P
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER NUMBER OF YEARS";T
40 I=I/100
50 PRINT"TOTAL INTEREST AMOUNT: $";
60 PRINT INT(P*I*T)
```

SAMPLE PROBLEM

If you deposit $1,000 at 8% simple interest for 5 years, how much interest would you have accumulated at the end of this period?

```
ENTER AMOUNT ? 1000
ENTER INTEREST RATE ? 8
ENTER NUMBER OF YEARS ? 5
TOTAL INTEREST AMOUNT: $ 400
```

COMPOUND INTEREST

Almost all lending and savings institutions involve working with compound interest. In a savings account with compound interest, you earn interest not only on the principal but also on the interest from the previous compounding period. In a borrowing transaction, you pay interest only on the unpaid balance, not on the original amount borrowed.

This program will prompt for the required variables and compute the compound interest:

```
10 INPUT"ENTER AMOUNT TO BE COMPOUNDED";P
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER NUMBER OF YEARS TO BE COMPOUNDED";T
40 INPUT"ENTER NUMBER OF COMPOUND PERIODS/YEAR";C
50 I=I/100
60 A=P*(1+I/C)↑(C*T)
70 PRINT"ACCUMULATED AMOUNT: $";
80 PRINT INT(A*100+.5)/100
```

SAMPLE PROBLEM

If you deposit $1,000 at 8% interest compounded quarterly for 5 years, how much money would you have at the end of this period?

```
ENTER AMOUNT TO BE COMPOUNDED ? 1000
ENTER INTEREST RATE ? 8
ENTER NUMBER OF YEARS TO BE COMPOUNDED ? 5
ENTER NUMBER OF COMPOUND PERIODS/YEAR ? 4
ACCUMULATED AMOUNT: $ 1485.95
```

BEST ANNUAL NOMINAL INTEREST RATE

When financial agencies quote an interest rate, it is usually specified as an annual rate with daily, monthly, quarterly, or some other periodic compounding period. In certain cases it is hard to determine which offering provides the best return and what the differences among them are. The following program will determine this:

```
10 INPUT"ENTER AMOUNT TO BE DEPOSITED";P
20 INPUT"ENTER NUMBER OF YEARS TO BE DEPOSITED";T
30 INPUT"ENTER INTEREST RATE:1";I1
40 INPUT"ENTER # OF COMPOUND PERIODS/YEAR:1";C1
50 INPUT"ENTER INTEREST RATE:2";I2
60 INPUT"ENTER # OF COMPOUND PERIODS/YEAR:2";C2
70 I1=I1/100:I2=I2/100
80 A1=P*(1+I1/C1)↑(C1*T)
90 A2=P*(1+I2/C2)↑(C2*T)
100 PRINT"INVESTMENT:1 RETURNS: $";
110 PRINT INT(A1*100+.5)/100
120 PRINT"INVESTMENT:2 RETURNS: $";
130 PRINT INT(A2*100+.5)/100
```

SAMPLE PROBLEM

Consider one agency that has an annual percentage interest rate of 6.5% compounded quarterly and another agency that has a rate of 6.0% compounded monthly. If $5,000 is to be deposited for 5 years, what will be the resulting amount from each agency?

```
ENTER AMOUNT TO BE DEPOSITED ? 5000
ENTER NUMBER OF YEARS TO BE DEPOSITED ? 5
ENTER INTEREST RATE:1 ? 6.5
ENTER # OF COMPOUND PERIODS/YEAR:1 ? 4
ENTER INTEREST RATE:2 ? 6.0
ENTER # OF COMPOUND PERIODS/YEAR:2 ? 12
INVESTMENT:1 RETURNS: $ 6902.10
INVESTMENT:2 RETURNS: $ 6744.25
```

NOMINAL RATE TO ANNUAL EFFECTIVE INTEREST RATE

A nominal yearly rate is that rate quoted by a financial institution. An annual effective interest rate can be determined by considering the compounded portion of the investment.

The following program will calculate the annual effective interest rate converted from the nominal rate.

```
10 INPUT"ENTER NOMINAL INTEREST RATE";I
20 INPUT"ENTER NUMBER OF COMPOUND PERIODS/YEAR";C
30 I=I/100
40 PRINT"EFFECTIVE INTEREST RATE:";
50 X=100*((1+I/C)↑C-1)
60 PRINT INT(X*100+.5)/100
```

SAMPLE PROBLEM

If a bank quotes a nominal interest rate of 7.5% compounded quarterly, what is the effective interest rate?

```
ENTER NOMINAL INTEREST RATE ? 7.5
ENTER NUMBER OF COMPOUND PERIODS/YEAR ? 4
EFFECTIVE INTEREST RATE: 7.71
```

ANNUITY PAYMENTS

Up to now we've only looked at situations in which one deposit was made. With an annuity, a series of equal payments is made at specific intervals. There are two types of annuities. With an ordinary annuity, the payments are made at the end of each payment period. The majority of loans are of this type. With an annuity due, the payments are made at the beginning of each period. Most leases are annuities due.

The following program will calculate the required payment amount of an ordinary annuity.

```
10 INPUT"ENTER AMOUNT OF LOAN";P
20 INPUT"ENTER ANNUAL INTEREST RATE";I
30 INPUT"ENTER NUMBER OF MONTHS OF REPAY LOAN";M
40 I=I/100
50 X=P*(I/12)/(1-(1+(I/12))↑-M)
60 X=INT(X*100+.5)/100
70 PRINT"REQUIRED MONTHLY PAYMENT IS: $";
80 PRINT X
```

SAMPLE PROBLEM

If you borrowed $3,000 for 4 years (48 months) at 12% interest, how much should the monthly annuity payment be?

```
ENTER AMOUNT OF LOAN ? 3000
ENTER ANNUAL INTEREST RATE ? 12
ENTER NUMBER OF MONTHS TO REPAY LOAN ? 48
REQUIRED MONTHLY PAYMENT IS: $ 79.00
```

FUTURE VALUE OF ANNUITY PAYMENTS

Sometimes individuals make monthly deposits as part of their personal savings effort. This is often done through company savings programs. The following program will calculate the future value of these annuity payments.

```
10 INPUT"ENTER AMOUNT OF MONTHLY DEPOSIT";P
20 INPUT"ENTER ANNUAL INTEREST RATE";I
30 INPUT"ENTER NUMBER OF MONTHS SAVED";M
40 I=I/1200
50 FV=P*(1+I)*(((1+I)↑M-1)/I)
60 FV=INT(FV*100+.5)/100
70 PRINT"FUTURE VALUE: $";FV
```

SAMPLE PROBLEM

If you deposit $150 a month into a savings account paying 8% annual interest, how much money would you have after 5 years (60 months)?

```
ENTER AMOUNT OF MONTHLY DEPOSIT ? 150
ENTER ANNUAL INTEREST RATE ? 8
ENTER NUMBER OF MONTHS SAVED ? 60
FUTURE VALUE: $ 11095.01
```

RATE OF RETURN REQUIRED TO OBTAIN SPECIFIC AMOUNT

In certain financial situations it's good to know what rate of return is required to obtain a specific amount. You need to specify how much you will start off with, how long you will invest it, what the compound periods per year are, and how much you wish to end up with. The following program will do this calculation:

```
10 INPUT"ENTER INITIAL AMOUNT";P
20 INPUT"ENTER FINAL AMOUNT DESIRED";A
30 INPUT"ENTER NUMBER OF YEARS INVESTED";T
```

```
40 INPUT"ENTER NUMBER OF COMPOUND PERIODS/YEAR";C
50 I=((A/P)↑(1/(C*T))-1)*C
60 I=I*100
70 I=INT(I*100+.5)/100
80 PRINT"REQUIRED RATE OF RETURN:";I
```

SAMPLE PROBLEM

You make an initial deposit of $10,000 in a certain investment and it will yield $50,000 in 15 years. If the money is compounded monthly, what is the yearly rate of return?

```
ENTER INITIAL AMOUNT ? 10000
ENTER FINAL AMOUNT DESIRED ? 50000
ENTER NUMBER OF YEARS INVESTED ? 15
ENTER NUMBER OF COMPOUND PERIODS/YEAR ? 12
REQUIRED RATE OF RETURN: 10.78
```

CONVERTING INTEREST RATES WITH DIFFERENT COMPOUNDING ON SINGLE DEPOSIT

Continuous compounding has recently become very popular with financial institutions. This method assumes that compounding occurs continuously over the time periods rather than at specific intervals. With specific compounding, a certain time period such as monthly or quarterly is used to calculate the amount of interest.

In certain situations it is helpful to convert a rate from one method to another. The following program converts an annual rate with continuous compounding to an equivalent specific rate:

```
10 INPUT"ENTER RATE WITH CONT. COMPOUNDING";I
20 I=I/100
30 X=EXP(I)-1
40 X=X*100
50 PRINT"EQUIVALENT SPECIFIC RATE:";
60 PRINT INT(X*100+.5)/100
```

To convert a specific rate to an equivalent continuous rate, the following program should be used:

```
10 INPUT"ENTER RATE WITH SPECIFIC COMPOUNDING:";I
20 I=I/100
30 X=LOG(1+I)
40 X=X*100
```

```
50 PRINT"EQUIVALENT CONTINUOUS RATE:";
60 PRINT INT(X*100+.5)/100
```

SAMPLE PROBLEM

The following sample problem converts an annual rate with continuous compounding to a specific rate.

```
ENTER RATE WITH CONT. COMPOUNDING ? 6.0
EQUIVALENT SPECIFIC RATE: 6.18
```

A sample problem that converts an annual rate with specific compounding to a continuous rate is:

```
ENTER RATE WITH SPECIFIC COMPOUNDING ? 6.0
EQUIVALENT CONTINUOUS RATE: 5.83
```

CONVERTING INTEREST RATES WITH DIFFERENT COMPOUNDING ON ANNUITY

As previously explained, with an annuity an individual deposits a certain amount of money and then receives a specific amount at fixed time intervals for a period of time. An annuity can also be set up such that an individual pays a certain amount at fixed time intervals for a specific amount of time and at the end of this time period receives a lump sum payment.

The following program converts an annual rate with continuous compounding to an equivalent specific rate for an annuity. This is with 12 monthly payments per year.

```
10 INPUT"ENTER RATE WTH CONT. COMPOUNDING";I
20 I=I/100
30 X=1200*(EXP(I/12)-1)
40 PRINT"EQUIVALENT SPECIFIC RATE:";
50 PRINT INT(X*100+.5)/100
```

To convert a specific rate to an equivalent continuous rate, the following program should be used:

```
10 INPUT"ENTER RATE WITH SPECIFIC COMPOUNDING";I
20 I=I/1200
30 X=1200*(LOG(1+I))
40 PRINT"EQUIVALENT CONTINUOUS RATE:";
50 PRINT INT(X*100+.5)/100
```

SAMPLE PROBLEM

The following sample problem converts an annual rate for an annuity with continuous compounding to a specific rate.

```
ENTER RATE WITH CONT. COMPOUNDING ? 7.0
EQUIVALENT SPECIFIC RATE: 7.02
```

A sample problem that converts an annual rate for an annuity with specific compounding to a continuous rate is:

```
ENTER RATE WITH SPECIFIC COMPOUNDING ? 7.0
EQUIVALENT CONTINUOUS RATE: 6.98
```

DEPOSIT REQUIRED TO OBTAIN SPECIFIC FUTURE AMOUNT WITH CONTINUOUS COMPOUNDING

This calculation is very useful when a specific lump sum will be needed by a certain time in the future. It deals with regular monthly deposits of a fixed amount. The following program will determine the deposit required:

```
10 INPUT"ENTER ACCUMULATED AMOUNT DESIRED";A
20 INPUT"ENTER NUMBER OF MONTHLY DEPOSITS";D
30 INPUT"ENTER ANNUAL INTEREST RATE";I
40 I=I/1200
50 I=LOG(1+I)
60 X=(A*I)/((1+I)↑(D+1)-(1+I))
70 PRINT"DEPOSIT REQUIRED: $";
80 PRINT INT(X*100+.5)/100
```

SAMPLE PROBLEM

If you would like to have $10,000 in 3 years (36 months) and the annual interest rate is 6% compounded continuously, what would be the monthly deposit required?

```
ENTER ACCUMULATED AMOUNT DESIRED ? 10000
ENTER NUMBER OF MONTHLY DEPOSITS ? 36
ENTER ANNUAL INTEREST RATE ? 6
DEPOSIT REQUIRED: $ 253.01
```

CHAPTER

2

Depreciation Method

When investing capital in equipment, buildings, automobiles, and other items related to your business, you can depreciate these assets and deduct them from your taxable income as regulated by the United States government. It is recognized that the value of these assets decreases for numerous reasons and that the government wants to encourage capital investments.

Depreciation is the estimate of the amount of market value decrease of an item over a specific time period. The useful life of this item is termed the expected life, and this is the length of time over which it should be depreciated in most instances. For some assets, the government provides guidelines for the expected life-time period. The total cost of the asset is depreciable over the expected life unless it has some value at the end of this period. The value of the asset at the end of the depreciable period is the salvage value, which can be determined by looking at the current market rate of comparable products or government published valuations of used equipment. At any time during the depreciation period, a book value of the asset can be determined by subtracting the accumulated depreciation from the cost.

There are four basic methods for calculating the depreciation of an asset—straight line, sum of the years' digits, double declining balance, and units of production. There are also a few other methods allowed by the government, but these are the most popular and, therefore, the ones illustrated here.

The equation used to determine the straight line depreciation is as follows:

3 1833 04371 1156

$$DEPR = \frac{C - S}{L}$$

where: DEPR = *yearly straight line depreciation*
C = *cost*
S = *salvage value*
L = *life of asset*

The equation used to determine the sum of the years' digits depreciation for year N is as follows:

$$DEPR = \left(\frac{C - S}{SYD}\right) \times (L + 1 - N)$$

where: DEPR = *sum of years' digits depreciation for year N*
C = *cost*
S = *salvage value*
$SYD = \frac{L \times (L + 1)}{2}$
L = *life of asset (years)*
N = *year for which depreciation is calculated*

The equation used to determine the double declining balance depreciation is as follows:

$$DEPR = BV \times \left(\frac{2}{L}\right)$$

where: DEPR = *double declining balance depreciation*
BV = *current book value (this cannot go below the salvage value)*
L = *life of asset*

The equation used to determine the units of production depreciation is as follows:

$$DEPR = \frac{C - S}{EUP}$$

where: DEPR = *depreciation expense per unit of production*
C = *cost*
S = *salvage value*
EUP = *estimated units of production over life of asset*

STRAIGHT LINE

The straight line method is the easiest to use and understand. The original cost or value of the asset is assumed to be equally and evenly distributed over its expected life. The yearly depreciation is calculated by dividing the total depreciable cost by the number of years of the expected life. This produces an equal depreciation amount for each year.

This straight line depreciation program will prompt for the required variables and produce the yearly depreciation amounts along with the resulting book values. It allows for a partial year of depreciation. (Note that zero is not a valid entry for the number of remaining months of depreciation.)

```
10 INPUT"ENTER COST";C
20 INPUT"ENTER SALVAGE VALUE";SV
30 INPUT"ENTER LIFE OF ASSET (YEARS)";L
40 PRINT"ENTER # OF MONTHS REMAINING IN FIRST    YEAR
OF DEPRECIATION"
50 INPUT YR
60 X=C-SV
70 PRINT"TOTAL DEPRECIABLE VALUE: $";X
80 X=X/L
90 X=INT(X*100+.5)/100
100 X1=X*YR/12
110 X1=INT(X1*100+.5)/100
120 Y=1:BV=C
130 PRINT"YEARLY DEPRECIABLE AMOUNTS:"
140 PRINT" "
150 PRINT"YEAR  DEPRECIATION  BOOK
160 PRINT"      VALUE ($)     VALUE ($)"
170 PRINT Y;TAB(5)X1;
180 DT=DT+X1
190 BV=C-DT
200 PRINT TAB(19)BV
210 X1=X:Y=Y+1
220 IF Y=L+1 THEN 240
230 GOTO 170
240 IF YR=12 THEN END
250 X1=X*(12-YR)/12
260 X1=INT(X1*100+.5)/100
270 PRINT Y;TAB(5)X1;
280 DT=DT+X1
290 BV=C-DT
300 PRINT TAB(19)BV
```

SAMPLE PROBLEM

If you purchased a truck at a cost of $16,000 and it has an estimated life of 5 years and a salvage value of $1,000, determine the depreciation amounts and book value amounts for each year. It was purchased with 3 months remaining in the first year.

```
ENTER COST ? 16000
ENTER SALVAGE VALUE ? 1000
ENTER LIFE OF ASSET (YEARS) ? 5
ENTER NUMBER OF MONTHS REMAINING IN FIRST YEAR OF
DEPRECIATION ? 3
TOTAL DEPRECIABLE VALUE: $15000
YEARLY DEPRECIABLE AMOUNTS:
YEAR  DEPRECIATION  BOOK
      VALUE ($)     VALUE ($)
1     750           15250
2     3000          12250
3     3000          9250
4     3000          6250
5     3000          3250
6     2250          1000
```

SUM OF THE YEARS' DIGITS

The sum of the years' digits method is an accelerated depreciation method that allows for greater depreciation in the early portion of an asset's expected life and less in the later years. For each depreciation year, a decreasing fraction is multiplied by the amount to be depreciated. The denominator of this fraction is the sum of the years' digits, and the numerator is the number of years remaining to be depreciated.

This sum of the years' digits program will prompt for the required variables and produce the yearly depreciation amounts and the resulting book values.

```
10 INPUT"ENTER COST";C
20 INPUT"ENTER SALVAGE VALUE";SV
30 INPUT"ENTER LIFE OF ASSET (YEARS)";L
40 X=C-SV
50 PRINT"TOTAL DEPRECIABLE VALUE: $";X
60 B=((L*L)+L)/2:N=1
70 PRINT"YEARLY DEPRECIABLE AMOUNTS:"
```

```
80 PRINT
90 PRINT"YEAR  DEPRECIATION  BOOK"
100 PRINT"      VALUE ($)     VALUE ($)"
110 BV=C
120 FOR Y=L TO 1 STEP -1
130 DP=(Y/B)*(C-SV)
140 DP=INT(DP*100+.5)/100
150 PRINT N;TAB(5)DP;
160 DT=DT+DP
170 BV=C-DT
180 BV=INT(BV*100+.5)/100
190 PRINT TAB(19)BV
200 N=N+1
210 NEXT Y
```

SAMPLE PROBLEM

If you purchased a machine at a cost of $30,000 and it has an estimated life of 10 years and a salvage value of $2,000, determine the depreciation amounts and book value amounts for each year.

```
ENTER COST ? 30000
ENTER SALVAGE VALUE ? 2000
ENTER LIFE OF ASSET (YEARS) ? 10
TOTAL DEPRECIABLE VALUE: $28000
YEARLY DEPRECIABLE AMOUNTS:
YEAR  DEPRECIATION  BOOK
      VALUE ($)     VALUE ($)
1     5090.91       24909.09
2     4581.82       20327.27
3     4072.73       16254.54
4     3563.64       12690.90
5     3054.55       9636.35
6     2545.45       7090.90
7     2036.36       5054.54
8     1527.27       3527.27
9     1018.18       2509.09
10    509.09        2000
```

DOUBLE DECLINING BALANCE

The double declining balance method is also an accelerated depreciation method that allows for a large amount of depreciation in the initial year and progressively smaller amounts in the later years. With this method, the depreciation amount is twice that of the straight line method. Each year this 200% straight line rate is multiplied by the previous year's book value, and the salvage value is not subtracted out. Depreciation stops when the book value reaches the salvage value, even if it is before the asset's determined expected life.

This double declining balance depreciation program will prompt for the required variables and produce the yearly depreciation amounts and the resulting book values.

```
10 INPUT"ENTER COST";C
20 INPUT"ENTER SALVAGE VALUE";SV
30 INPUT"ENTER LIFE OF ASSET (YEARS)";L
40 X=C-SV
50 PRINT"TOTAL DEPRECIABLE VALUE: $";X
60 BV=C:N=1
70 PRINT"YEARLY DEPRECIABLE AMOUNTS:"
80 PRINT
90 PRINT"YEAR  DEPRECIATION  BOOK"
100 PRINT"      VALUE ($)     VALUE ($)"
110 FOR Y=L TO 1 STEP -1
120 DP=2*(BV/L)
130 DT=DT+DP
140 BV=C-DT
150 IF BV<SV THEN 230
160 DP=INT(DP*100+.5)/100
170 BV=INT(BV*100+.5)/100
180 PRINT N;TAB(5)DP;
190 PRINT TAB(19)BV
200 N=N+1
210 NEXT Y
220 END
230 Z=SV-BV
240 DP=DP-Z
250 BV=SV:Y=1
260 GOTO 160
```

SAMPLE PROBLEM

If you purchased a building at a cost of $200,000 and it has an estimated life of 15 years and a salvage value of $25,000, determine the depreciation amounts and the book value amounts for each year.

```
ENTER COST ? 200000
ENTER SALVAGE VALUE ? 25000
```

```
ENTER LIFE OF ASSET (YEARS) ? 15
TOTAL DEPRECIABLE VALUE: $175000
YEARLY DEPRECIABLE AMOUNTS:
YEAR  DEPRECIATION  BOOK
      VALUE ($)     VALUE ($)
1     26666.67      173333.33
2     23111.11      150222.22
3     20029.63      130192.59
4     17359.01      112833.58
5     15044.48      97789.10
6     13038.55      84750.56
7     11300.07      73450.48
8     9793.40       63657.09
9     8487.61       55169.47
10    7355.93       47813.54
11    6375.14       41438.41
12    5525.12       35913.28
13    4788.44       31124.85
14    4149.98       26974.87
15    1974.87       25000
```

UNITS OF PRODUCTION

The units-of-production method depreciates an asset according to how much it is used and not by a specific period of time. The depreciation per unit is found by dividing the depreciable cost by the expected number of units it can produce. The yearly depreciation amount is then found by multiplying this per-unit cost by the number of units produced.

This units-of-production program will prompt for the required variables and produce the annual depreciation amount.

```
10 INPUT"ENTER COST";C
20 INPUT"ENTER SALVAGE VALUE";SV
30 INPUT"ENTER EXPECTED # OF UNITS PRODUCED";UN
40 INPUT"ENTER # OF UNITS PRODUCED DURING YEAR";UP
50 X=(C-SV)/UN
60 X=INT(X*100+.5)/100
70 PRINT"DEPRECIATION PER UNIT: $";X
80 Z=X*UP
90 PRINT"ANNUAL DEPRECIATION: $";Z
```

SAMPLE PROBLEM

If you purchase a machine at a cost of $5,000 and it will have a salvage value of $500 after it produces 10,000 units, what is the depreciation amount for a year in which 3,000 units were produced?

```
ENTER COST ? 5000
ENTER SALVAGE VALUE ? 500
ENTER EXPECTED # OF UNITS PRODUCED ? 10000
ENTER # OF UNITS PRODUCED DURING YEAR ? 3000
DEPRECIATION PER UNIT: $ .45
ANNUAL DEPRECIATION: $ 1350
```

CHAPTER

3

Retailing

In order for a retail firm to stay in business, it must sell its products for more than it pays for them. A merchant must increase the price to cover overhead (i.e., rent, utilities, wages, and taxes). The merchant is also motivated by a desire to make a profit above the overhead and merchandise cost. This overhead and profit are combined and called the markup. The markup is the difference between the cost of the product to the merchant and what the merchant sells it for.

The amount of money that a retailer pays for his merchandise is the cost price or wholesale price. The amount that he receives for his merchandise is the selling price or retail price. In a retailing environment the markup on goods can be based either on the cost or the selling price. Both of these methods are presented in this section. The markup can be expressed as an amount or a percent.

Sales tax computation and discounting are also presented in this section demonstrating their usage in the retailing environment.

SALES TAX AND PRICE

When a business sells merchandise to a customer, it must charge a sales tax on that product. The amount of this tax is a percentage of the selling price, which is set by each state and some city governments.

The following program computes the sales tax amount and the total price with regard to the initial price and a sales tax percent. It should be noted that the sales tax percent is a percentage value and should not be entered in its decimal format.

```
10 INPUT"ENTER PURCHASE PRICE";P
20 INPUT"ENTER SALES TAX RATE";TR
30 X=(TR/100)*P
40 X=INT(X*100+.5)/100
50 PRINT"SALES TAX: $";X
60 P=P+X
70 PRINT"TOTAL PRICE: $";P
```

SAMPLE PROBLEM

If you purchase a car at a cost of $9,500 and the sales tax is 6%, what is the amount of the tax and the total price?

```
ENTER PURCHASE PRICE ? 9500
ENTER SALES TAX RATE ? 6
SALES TAX: $ 570
TOTAL PRICE: $ 10070
```

PRICE BEFORE SALES TAX

Some companies quote a price for a specific piece of merchandise with the sales tax included. When comparing this product with a different retailer's product who does not add in the sales tax, it is useful to determine the initial sales price without the sales tax added.

The following program computes the sales tax, given the total price and the sales tax rate. It should be noted that the sales tax percent is a percentage value and should not be entered in its decimal format.

```
10 INPUT"ENTER TOTAL PRICE INCLUDING SALES TAX";TP
20 INPUT"ENTER SALES TAX RATE";TR
30 X=TP/(1+TR/100)
40 X=INT(X*100+.5)/100
50 PRINT"INITIAL PRICE: $";X
60 T=TP-X
70 T=INT(T*100+.5)/100
80 PRINT"SALES TAX: $";T
```

SAMPLE PROBLEM

If you purchase a telephone at a total cost of $96.35 and the sales tax rate is 5.25%, what is the initial price and sales tax amount?

```
ENTER TOTAL PRICE INCLUDING SALES TAX ? 96.35
ENTER SALES TAX RATE ? 5.25
INITIAL PRICE: $ 91.54
SALES TAX: $ 4.81
```

PURCHASE PRICE DISCOUNT

Numerous retailers have special sales throughout the year by discounting their merchandise. The items are usually discounted by a percentage of the retail price.

The following program computes the reduction amount and the reduced price with the specified discount percent.

```
10 INPUT"ENTER INITIAL PRICE";P
20 INPUT"ENTER DISCOUNT PERCENT";DP
30 X=P-(DP/100)*P
40 X=INT(X*100+.5)/100
50 PRINT"REDUCTION AMOUNT: $";P-X
60 PRINT"REDUCED PRICE WITH DISCOUNT: $";X
```

SAMPLE PROBLEM

If a store had a product with an initial price of $59.95 and it was discounted 25%, what would be the reduction amount and the final reduced price?

```
ENTER INITIAL PRICE ? 59.95
ENTER DISCOUNT PERCENT ? 25
REDUCTION AMOUNT: $ 14.99
REDUCED PRICE WITH DISCOUNT: $ 44.96
```

PRICE BEFORE MARKDOWN

Sometimes when retailers mark down merchandise, only the reduced price is displayed. In certain instances it is valuable to determine the initial price of the product. The discounted percent is needed to compute this calculation.

The following program computes the initial retail price prior to the markdown. The reduction amount is also given.

```
10 INPUT"ENTER PRICE AFTER MARKDOWN";P
20 INPUT"ENTER DISCOUNTED PERCENT";DP
30 X=P/(1-DP/100)
40 X=INT(X*100+.5)/100
50 PRINT"REDUCTION AMOUNT: $"X-P
60 PRINT"PRICE BEFORE DISCOUNT: $";X
```

SAMPLE PROBLEM

If a store is selling a product with a marked down price of $19.95 and it is discounted 30%, what was the initial price and the reduction amount?

```
ENTER PRICE AFTER MARKDOWN ? 19.95
ENTER DISCOUNTED PERCENT ? 30
REDUCTION AMOUNT: $ 8.55
PRICE BEFORE DISCOUNT: $ 28.50
```

COST BEFORE MARKUP

Many retailers have a standard markup percentage on their merchandise over the initial wholesale cost. In certain cases it is valuable to determine the cost before the retail markup, given this standard percentage.

The following program computes the cost before markup and the marked up amount.

```
10 INPUT"ENTER RETAIL PRICE";P
20 INPUT"ENTER MARKUP PERCENT";MP
30 X=P/(1+MP/100)
40 X=INT(X*100+.5)/100
50 PRINT"MARKUP AMOUNT: $";P-X
60 PRINT"COST BEFORE MARKUP: $";X
```

SAMPLE PROBLEM

If a store is selling a personal computer at a retail price of $99 and its markup percent is 45%, what is their cost before markup and the marked up amount?

```
ENTER RETAIL PRICE ? 99
ENTER MARKUP PERCENT ? 45
MARKUP AMOUNT: $ 30.72
COST BEFORE MARKUP: $ 68.28
```

MARKUP WITH SELLING PRICE

When a business purchases inventory to sell at retail, each item is marked up a certain percentage to allow the business to cover its expenses and make a profit.

The following program computes the retail price and the marked up amount, given the wholesale cost and the markup percentage.

```
10 INPUT"ENTER WHOLESALE COST";C
20 INPUT"ENTER MARKUP PERCENTAGE";MP
30 X=C*(1+MP/100)
40 X=INT(X*100+.5)/100
50 PRINT"MARKUP AMOUNT: $";X-C
60 PRINT"RETAIL PRICE: $";X
```

SAMPLE PROBLEM

If a retailer purchases a product for $250 and the markup percentage is 75%, what is the selling price and the marked up amount?

```
ENTER WHOLESALE COST ? 250
ENTER MARKUP PERCENTAGE ? 75
MARKUP AMOUNT: $ 187.50
RETAIL PRICE: $ 437.50
```

RETAIL PRICE TO EARN SPECIFIED GROSS PROFIT MARGIN

A retailer purchases his inventory at a wholesale cost and then increases this cost for his selling price. In the previous section we looked at a method whereby the retailer marked up his merchandise by a percentage of cost. In this section we will examine the Gross Profit Margin (GPM) method. With this method, the retailer increases the price to get a percentage profit margin on sales, not cost.

The following program computes the retail price and markup amount based on the retailer's required gross profit margin on sales.

```
10 INPUT"ENTER WHOLESALE COST";C
20 INPUT"ENTER GROSS PROFIT MARGIN DESIRED";PM
30 X=C/(1-PM/100)
40 X=INT(X*100+.5)/100
50 PRINT"MARKUP AMOUNT: $";X-C
60 PRINT"RETAIL PRICE: $";X
```

SAMPLE PROBLEM

If a retailer purchases a product for $250 and requires a gross profit margin on sales of 30%, what is the selling price and the marked up amount?

```
ENTER WHOLESALE COST ? 250
ENTER GROSS PROFIT MARGIN DESIRED ? 30
MARKUP AMOUNT: $ 107.14
RETAIL PRICE: $357.14
```

PERCENT MARKUP ON COST

Many manufacturing companies give a suggested retail price for their products and sell their products to the retailer at a wholesale price. In most situations the retailer needs to determine what percent markup the suggested retail price is.

The following program computes the percentage markup on cost, given the wholesale cost and the retail price.

```
10 INPUT"ENTER RETAIL PRICE";P
20 INPUT"ENTER WHOLESALE COST";C
30 X=(P/C)-1
40 X=INT(X*100+.5)/100
50 PRINT"PERCENTAGE MARKUP ON COST:";X*100
```

SAMPLE PROBLEM

If a retailer purchased a piece of furniture for $350 and it has a selling price of $600, what is the percentage markup on cost?

```
ENTER RETAIL PRICE ? 600
ENTER WHOLESALE COST ? 350
PERCENTAGE MARKUP ON COST: 71
```

PERCENT MARKUP ON SALES

In the previous section we looked at a percentage markup on cost. In this section we are looking at a pecentage markup on sales. As stated in the previous section, this allows the retailer to determine the markup percent when given a wholesale cost and a retail price.

The following program computes the percentage markup on sales given the wholesale cost and the retail price.

```
10 INPUT"ENTER RETAIL PRICE";P
20 INPUT"ENTER WHOLESALE COST";C
30 X=1-(C/P)
40 X=INT(X*100+.5)/100
50 PRINT"PERCENTAGE MARKUP ON SALES:";X*100
```

SAMPLE PROBLEM

If a retailer purchased a piece of machinery for $2,500 and it has a selling price of $3,200, what is the percentage markup on sales?

```
ENTER RETAIL PRICE ? 3200
ENTER WHOLESALE COST ? 2500
PERCENTAGE MARKUP ON SALES: 22
```

RETAILER'S COST BASED ON GROSS PROFIT MARGIN

In certain situations where a suggested retail price must be used, the retailer would like to determine the cost level to obtain a specific gross profit margin (GPM).

The following program computes the wholesale cost and the markup amount, given a retail price and a desired gross profit margin on the selling price.

```
10 INPUT"ENTER RETAIL PRICE";P
20 INPUT"ENTER PROFIT MARGIN ON SELLING PRICE";PM
30 X=P*(1-PM/100)
```

```
40 X=INT(X*100+.5)/100
50 PRINT"MARKUP AMOUNT: $";P-X
60 PRINT"WHOLESALE COST: $";X
```

SAMPLE PROBLEM

If a retail price for a book is $14.95 and the retailer wishes to have a 25% profit margin on sales, what must his cost be for the book?

```
ENTER RETAIL PRICE ? 14.95
ENTER PROFIT MARGIN ON SELLING PRICE ? 25
MARKUP AMOUNT: $ 3.74
WHOLESALE COST: $ 11.21
```

MARKUP ON SALES VERSUS COST

In numerous situations you have to determine the wholesale cost, given a retail price and a markup percent. This markup increase may be based on the selling price or cost. In this section you are given a mechanism to compare the two markup methods.

The following program computes the wholesale cost based on a markup of sales and a markup on cost.

```
10 INPUT"ENTER RETAIL PRICE";P
20 INPUT"ENTER % MARKUP ON SALES";MS
30 INPUT"ENTER % MARKUP ON COST";MC
40 X1=P*(1-MS/100)
50 X2=P/(1+MC/100)
60 X1=INT(X1*100+.5)/100
70 X2=INT(X2*100+.5)/100
80 PRINT"WHOLESALE COST WITH MARKUP ON SALES: $";X1
90 PRINT"WHOLESALE COST WITH MARKUP ON COST: $";X2
```

SAMPLE PROBLEM

You have the chance to buy a television from two different stores at cost. This product has a normal retail price of $359. One of the stores has a markup on cost of 30%, and the other has a markup on sales of 25%. Which store would you buy from?

```
ENTER RETAIL PRICE ? 359
ENTER % MARKUP ON SALES ? 25
```

```
ENTER % MARKUP ON COST ? 30
WHOLESALE COST WITH MARKUP ON SALES: $ 269.25
WHOLESALE COST WITH MARKUP ON COST: $ 276.15
```

PERCENT OF MARKDOWN

When considering a purchase where the retailer gives an original price and a sales price, it is sometimes beneficial to determine the percent of markdown. This is useful in determining which item has been marked down the most and is, therefore, the best buy.

The following program computes the percent markdown, given the original price and sales price.

```
10 INPUT"ENTER ORIGINAL PRICE";IP
20 INPUT"ENTER SALES PRICE";SP
30 X=(1-(SP/IP))*100
40 X=INT(X*100+.5)/100
50 PRINT"PERCENT MARKDOWN: ";X
```

SAMPLE PROBLEM

If a dress had an original price of $85 and is marked down to $63, what is the percent of the markdown?

```
ENTER ORIGINAL PRICE ? 85
ENTER SALES PRICE ? 63
PERCENT MARKDOWN: 25.88
```

EARLY PAYMENT DISCOUNT

In the business world many transactions allow for a reduced price for early payment. This early payment discount is usually quoted in the following format: 2%,10; net 30. This means a 2% discount can be taken if paid within 10 days, and if not, the full amount is due in 30 days.

The following program computes the early price, given the regular price and the early payment discount.

```
10 INPUT"ENTER REGULAR PRICE";RP
20 INPUT"ENTER EARLY PAYMENT DISCOUNT %";D
```

```
30 DP=RP-(RP*D/100)
40 DP=INT(DP*100+.5)/100
50 PRINT"EARLY PAYMENT PRICE: $";DP
```

SAMPLE PROBLEM

If you purchase an item for $2,000 with terms of 2%,10; net 30, what would be the early payment price if you paid it off within 10 days?

```
ENTER REGULAR PRICE ? 2000
ENTER EARLY PAYMENT DISCOUNT % ? 2
EARLY PAYMENT PRICE: $ 1960
```

COMPARING QUANTITY PURCHASE DISCOUNTS

Retailers use different types of strategy when pricing similar items. One of the biggest differences is quantity discounts, but it's not always clear which offering has the lowest price per unit.

```
10 INPUT"ENTER PRICE FOR SALE-1";P1
20 INPUT"ENTER QUANTITY FOR SALE-1";Q1
30 INPUT"ENTER PRICE FOR SALE-2";P2
40 INPUT"ENTER QUANTITY FOR SALE-2";Q2
50 X1=P1/Q1
60 X2=P2/Q2
70 PRINT"UNIT PRICE SALE-1: $";X1
80 PRINT"UNIT PRICE SALE-2: $";X2
```

SAMPLE PROBLEM

If one company offered to sell you 5,000 pencils for $60, and another company offered to sell you 3,000 pencils for $35, which company has the lowest per-unit price?

```
ENTER PRICE FOR SALE-1 ? 60
ENTER QUANTITY FOR SALE-1 ? 5000
ENTER PRICE FOR SALE-2 ? 35
ENTER QUANTITY FOR SALE-2 ? 3000
UNIT PRICE SALE-1: $ .012
UNIT PRICE SALE-2: $ .01167
```

CHAPTER

4

Real Estate

For most people the purchase of their home is the largest financial undertaking they will ever attempt. Because real estate is such a large investment, all functions of the process must be analyzed carefully. The programs in this chapter demonstrate many areas of this analysis process.

When purchasing real estate most people take out a mortgage loan in order to buy the property. When a piece of property is mortgaged, the buyer must repay the loan with a monthly payment of principal and interest. The mortgage company may also require that money for taxes and insurance be included with the monthly payment. In the programs presented in this chapter the monthly payments include money only for principal and interest unless otherwise stated.

With the repayment of a standard real estate mortgage, all the monthly payments are equal unless there is a balloon payment at the end of the loan. The balloon payment allows for early payoff of the loan.

There are sometimes commissions and points paid when real estate is bought or sold. A commission is paid to the real estate agency if it provides a service by selling your home. The seller pays the realtor's commission. Points can be paid by both the selling and buying parties. A point, which is one percent of the loan amount, is paid to the lending agency. Through points, the lending agency charges money to provide special low interest loans.

All interest rates specified in this chapter are annual percentage interest rates compounded annually unless otherwise stated.

TOTAL SQUARE FOOTAGE

In the real estate evaluation process, when some types of buildings are involved, the total square footage must be calculated. The total square footage can be determined roughly by taking the outside dimensions of the structure. A more exact calculation can be done by taking the dimensions of each room in the building. This is also useful when determining square footage in a portion of an office building or apartment.

This program accepts the dimensions of each room. Completion is signified by entering "0" for the first dimension of a room.

The following program computes the total square footage, given the dimensions of each room.

```
10 PRINT"ENTER ROOM DIMENSIONS"
20 PRINT"ENTER <0> TO FINISH ENTRIES"
30 INPUT"ENTER FIRST DIMENSION";D1
40 IF D1=0 THEN 100
50 INPUT"ENTER SECOND DIMENSION";D2
60 PRINT" "
70 X=D1*D2
80 T=T+X
90 GOTO 30
100 PRINT"TOTAL SQUARE FOOTAGE: ";T
```

SAMPLE PROBLEM

What is the total square footage of a house that has the following room sizes: 12×12, 14×12, 14×14, 25×20, 7×8, 6×9, 13×24, and 7×9?

```
ENTER ROOM DIMENSIONS
ENTER <0> TO FINISH ENTRIES
ENTER FIRST DIMENSION ? 12
ENTER SECOND DIMENSION ? 12
ENTER FIRST DIMENSION ? 14
ENTER SECOND DIMENSION ? 12
ENTER FIRST DIMENSION ? 14
ENTER SECOND DIMENSION ? 14
ENTER FIRST DIMENSION ? 25
ENTER SECOND DIMENSION ? 20
ENTER FIRST DIMENSION ? 7
ENTER SECOND DIMENSION ? 8
```

```
ENTER FIRST DIMENSION ? 6
ENTER SECOND DIMENSION ? 9
ENTER FIRST DIMENSION ? 13
ENTER SECOND DIMENSION ? 24
ENTER FIRST DIMENSION ? 7
ENTER SECOND DIMENSION ? 9
ENTER FIRST DIMENSION ? 0
TOTAL SQUARE FOOTAGE: 1493
```

COST PER SQUARE FOOT

When evaluating real estate one of the major items of concern is the cost per square foot. This is very important whether you are considering the purchase of an office building or a condominium on the lake. Its major use in the evaluation process is to compare it to other like properties.

The following program computes the cost per square foot, given the cost of the house and the total number of square feet that it contains.

```
10 INPUT"ENTER COST OF HOUSE";C
20 INPUT"ENTER NUMBER OF SQUARE FEET";SF
30 X=C/SF
40 X=INT(X*100+.5)/100
50 PRINT"COST PER SQUARE FOOT: $";X
```

SAMPLE PROBLEM

If you are considering the purchase of a home that has 2,400 square feet and is $130,000, what is the cost per square foot?

```
ENTER COST OF HOUSE ? 130000
ENTER NUMBER OF SQUARE FEET ? 2400
COST PER SQUARE FOOT: $ 54.17
```

DOWN PAYMENT REQUIRED WITH MORTGAGE ASSUMPTION

In some real estate transactions the buyer and seller agree that the buyer will assume the seller's mortgage on the property being sold. This is very popular with some of the older mortgages since they usually have substantially lower interest rates.

All mortgages are not assumable and the assumability is defined when the mortgage is originally written. Other types of mortgages allow assumption if specified penalties are paid.

The program shown here assumes that the down payment made to purchase the property will equal the equity that the owner has in the property. No penalties are paid with this mortgage assumption.

The following program computes the required down payment and the percentage of the down payment. Inputs to the program are the current value of the property and the balance due on the mortgage.

```
10 INPUT"ENTER VALUE OF PROPERTY";V
20 INPUT"ENTER AMOUNT STILL OWED ON PROPERTY";A
30 X=V-A
40 PRINT"REQUIRED DOWN PAYMENT: $";X
50 Y=(X/V)*100
60 Y=INT(Y*100+.5)/100
70 PRINT"PERCENT OF DOWN PAYMENT: ";Y
```

SAMPLE PROBLEM

If you are selling your home for $90,000 and you still owe $73,456 on your present mortgage, how much of a down payment must a buyer put down to assume your mortgage?

```
ENTER VALUE OF PROPERTY ? 90000
ENTER AMOUNT STILL OWED ON PROPERTY ? 73456
REQUIRED DOWN PAYMENT: $ 16544
PERCENT OF DOWN PAYMENT: 18.38
```

COMMISSIONS

When a real estate agency sells a piece of property, the owner of the property pays the agency a commission for finding the buyer and for preparing the paperwork to complete the transaction. This commission is a percentage of the selling price and is currently around 6%.

The commission is usually broken down into several portions for the different parties involved in the selling process. The program demonstrated in this section allows a percentage for the real estate firm, the selling agent, and the listing agent. Some real estate firms may break the commission down into more portions, but this is the most popular practice.

The following program computes the total commission paid, the real estate agent's portion, the listing agent's portion, and the real estate firm's portion. The inputs to this program are the amount of the sale and the percentage of the commission for each party involved.

```
10 INPUT"ENTER TOTAL COMMISSION %";TC
20 INPUT"ENTER % FOR REAL ESTATE FIRM";FC
30 INPUT"ENTER % FOR SELLING AGENT";AC
40 INPUT"ENTER % FOR LISTING AGENT";LC
50 INPUT"ENTER AMOUNT OF SALE";S
60 IF TC=FC+AC+LC THEN 90
70 PRINT"COMMISSIONS DO NOT ADD UP CORRECTLY"
80 END
90 TC=INT((TC/100*S)*100+.5)/100
100 PRINT"TOTAL COMMISSION: $";TC
110 AC=INT((AC/100*S)*100+.5)/100
120 PRINT"SELLING AGENTS COMMISSION: $";AC
130 LC=INT((LC/100*S)*100+.5)/100
140 PRINT"LISTING AGENTS COMMISSION: $";LC
150 FC=INT((FC/100*S)*100+.5)/100
160 PRINT"REAL ESTATE FIRMS COMMISSION: $";FC
```

SAMPLE PROBLEM

A real estate agency that charges a 6% commission for selling a house breaks down the commission as 1% for the listing agent, 2% for the selling agent, and 3% for the real estate firm. How much will each of these parties receive if a $185,000 house is sold?

```
ENTER TOTAL COMMISSION % ? 6
ENTER % FOR REAL ESTATE FIRM ? 3
ENTER % FOR SELLING AGENT ? 2
ENTER % FOR LISTING AGENT ? 1
ENTER AMOUNT OF SALE ? 185000
TOTAL COMMISSION: $ 11100
SELLING AGENTS COMMISSION: $ 3700
LISTING AGENTS COMMISSION: $ 1850
REAL ESTATE FIRMS COMMISSION: $ 5550
```

APPRECIATION RATE

In the last ten years inflation has caused the value of real estate to escalate at a very rapid rate. The program shown in this section allows you to determine the average yearly appreciation rate of a piece of property. This computation is an iterative process so it may take a few seconds to produce the result.

The following program computes the average yearly appreciation rate, given the initial cost, the present valuation, and the number of years of the appreciation period.

```
10 INPUT"ENTER INITIAL COST OF PROPERTY";IC
20 INPUT"ENTER PRESENT VALUE OF PROPERTY";PV
30 INPUT"ENTER APPRECIATION PERIOD (YEARS)";N
40 S=.005:BG=.0001
50 FOR I=BG TO 10 STEP S
60 X=PV*(1+I)↑-N
70 IF IC>=X THEN 90
80 NEXT I
90 IF S=.0001 THEN 120
100 S=.0001:BG=I-.005
110 GOTO 50
120 I=I*100
130 I=INT(I*100+.5)/100
140 PRINT"AVERAGE YEARLY APPRECIATION RATE: ";I
```

SAMPLE PROBLEM

If you purchased a home 6 years ago for $35,450 and it is currently worth $72,500, what has been the average yearly appreciation rate?

```
ENTER INITIAL COST OF PROPERTY ? 35450
ENTER PRESENT VALUE OF PROPERTY ? 72500
ENTER APPRECIATION PERIOD (YEARS) ? 6
AVERAGE YEARLY APPRCIATION RATE : 12.67
```

FUTURE VALUE OF REAL ESTATE

When you invest in anything that is expected to appreciate in value, it is always interesting to determine what the future worth will be. This is also important if you are planning to sell when your property reaches a certain value.

The following program computes the future value of a piece of real estate, given the cost of the property, the appreciation rate, and the number of years to display.

```
10 INPUT"ENTER COST OF PROPERTY";PV
20 INPUT"ENTER APPRECIATION RATE";I
30 INPUT"ENTER # OF YEARS TO DISPLAY";YR
40 I=I/100
50 PRINT"YEAR  VALUE"
60 FOR N=1 TO YR
70 FV=PV*(1+I)↑N
80 FV=INT(FV*100+.5)/100
90 PRINT N;TAB(6)FV
100 NEXT N
```

SAMPLE PROBLEM

If you purchased a home for $43,250 and expect it to appreciate at 7% per year, what will be the future value each year over the next 10 years?

```
ENTER COST OF PROPERTY ? 43250
ENTER APPRECIATION RATE ? 7
ENTER # OF YEARS TO DISPLAY ? 10
YEAR  VALUE
1     46277.50
2     49516.92
3     52983.11
4     56691.93
5     60660.36
6     64906.59
7     69450.05
8     74311.55
9     79513.36
10    85079.30
```

EFFECTIVE INTEREST RATE WHEN POINTS ARE PAID

Some lending institutions offer loans with very appealing interest rates if the buyer will pay a certain number of points. Each point is one percent of the loan amount. When the loan is made, the buyer must pay the lending agency the specified percentage with regard to the number of points required.

Since the person borrowing the money is actually paying money to get a specific loan, he is in effect receiving less money but paying back the full amount. The program in this section computes the effective interest rate by considering the balance borrowed as the loan amount less the amount paid for points. This computation is an interative process so it may take a few seconds to produce the result.

The following program computes the effective interest rate on a loan when the buyer pays points. The inputs to the program are the loan amount, the time period of the loan in months, the monthly payment, and the number of points that the buyer pays.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
30 INPUT"ENTER MONTHLY PAYMENT";PM
40 INPUT"ENTER # OF POINTS BUYER PAYS";P
50 A=A*(1-P/100)
60 X=A/PM
70 S=.001:BG=.0008
80 FOR I=BG TO .1 STEP S
90 Z=(1-(1+I)↑-N)/I
100 IF Z>X THEN NEXT I
110 IF S<>.001 THEN 140
120 S=.00001:BG=I-.001
130 GOTO 80
140 I=I*1200
150 I=INT(I*100+.5)/100
160 PRINT"EFFECTIVE INTEREST RATE: ";I
```

SAMPLE PROBLEM

You purchased a home which required you to take out a loan of $67,500 for a 30-year (360-month) period, and your monthly payment for principal and interest is $690. What is the effective interest rate of this mortgage if 5 points had to be paid by the buyer?

```
ENTER LOAN AMOUNT ? 67500
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
ENTER MONTHLY PAYMENT ? 690
ENTER # OF POINTS BUYER PAYS ? 5
EFFECTIVE INTEREST RATE: 12.62
```

SELLING PRICE WHEN SELLER PAYS POINTS

In certain situations the seller of a piece of property will pay points to a lending agency to enable it to offer a lower interest rate to potential buyers. Many home builders do this when interest rates rise extemely high. In many cases this is an artificial enticement, because the amount the builder pays for points is usually added to the selling price of the home.

Each point is 1% of the loan amount. When the loan is made, the seller must pay the lending agency the specified percentage with regard to the number of points required.

The program presented in this section determines the selling price to allow the seller to make a specified amount of profit, taking into account that the seller must pay points. Many builders and land developers use this type of calculation to acquire the return they desire.

The following program computes the selling price, given the amount paid for the house, the profit desired, and the number of points the seller must pay.

```
10 INPUT"ENTER AMOUNT PAID FOR HOUSE";A
20 INPUT"ENTER PROFIT YOU DESIRE";PR
30 INPUT"ENTER # OF POINTS SELLER PAYS";P
40 X=(A+PR)/(1-P/100)
50 X=INT(X*100+.5)/100
60 PRINT"SELLING PRICE: $";X
```

SAMPLE PROBLEM

In order to sell your home you are going to have to pay 3 points. If you paid $71,750 for your home and you wish to make a $20,000 profit, what must the selling price be to attain your desired profit?

```
ENTER AMOUNT PAID FOR HOUSE ? 71750
ENTER PROFIT YOU DESIRE ? 20000
ENTER # OF POINTS SELLER PAYS ? 3
SELLING PRICE: $ 94587.63
```

SELLER'S PROFIT WHEN SELLER PAYS POINTS AND COMMISSION

When individuals sell property, they usually have to pay a real estate agency a commission for selling it for them and, depending on the economic times and the loans available, they may also have to pay points. The points paid by the seller allow the buyer to get lower interest rates and thus help in the sale of the property.

Each point is 1% of the loan amount. When the loan is made, the seller must pay the lending agency the specified percentage with regard to the number of points required.

The following program computes the amount the seller receives when he has to pay a realtor's commission and has to pay points. The inputs to this program are the selling price, the percent of commission for the real estate agency, and the number of points the seller must pay.

```
10 INPUT"ENTER SELLING PRICE";SP
20 INPUT"ENTER % COMMISSION PAID TO REALTOR";C
30 INPUT"ENTER # OF POINTS SELLER PAYS";P
40 X=SP*(C/100)
50 X=INT(X*100+.5)/100
60 PRINT"REALTORS PORTION: $";X
70 Y=SP*(P/100)
80 Y=INT(Y*100+.5)/100
90 PRINT"PORTION PAID FOR POINTS: $";Y
100 SP=SP-(X+Y)
110 PRINT"AMOUNT SELLER RECEIVES: $";SP
```

SAMPLE PROBLEM

If you sold your home for $115,000 and paid a real estate firm a 6% commission, how much would you receive if you had to pay 4 points?

```
ENTER SELLING PRICE ? 115000
ENTER % COMMISSION PAID TO REALTOR ? 6
ENTER # OF POINTS SELLER PAYS ? 4
REALTORS PORTION: $ 6900
PORTION PAID FOR POINTS: $ 4600
AMOUNT SELLER RECEIVES: $ 103500
```

MONTHLY MORTGAGE PAYMENT

Some monthly mortgage payments include an amount for principal and interest and also an amount for taxes and insurance. The principal and interest go toward paying off the loan, and the taxes and insurance portion goes into an escrow account. From this escrow account the mortgage company pays the taxes and insurance when they are due.

The following program computes the portion of the payment for principal and interest and also the total mortgage payment. The inputs to this program are the loan amount, the annual interest rate, the time period of the loan in months, and the percent of the loan payment for taxes and insurance.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER ANNUAL INTEREST RATE";I
30 INPUT"ENTER TIME  PERIOD OF LOAN (MONTHS)";N
40 INPUT"ENTER % OF LOAN PAYMENT FOR TAX & INS";T
50 I=I/1200
60 PM=A/((1-(1+I)↑-N)/I)
70 PM=INT(PM*100+.5)/100
80 PRINT"PAYMENT FOR PRINCIPAL AND INTEREST: $";PM
90 PM=PM*(1+T/100)
100 PM=INT(PM*100+.5)/100
110 PRINT"TOTAL MONTHLY PAYMENT: $";PM
```

SAMPLE PROBLEM

If you purchased a home that required you to take out a $60,000 mortgage at 12% interest over a 30-year (360-month) period, what would be the total monthly payment if 25% of the payment was required for taxes and insurance?

```
ENTER LOAN AMOUNT ? 60000
ENTER ANNUAL INTEREST RATE ? 12
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
ENTER % OF LOAN PAYMENT FOR TAX & INS ? 25
PAYMENT FOR PRINCIPAL AND INTEREST: $ 617.17
TOTAL MONTHLY PAYMENT: $771.46
```

NEW MONTHLY PAYMENT WHEN REFINANCING

When interest rates fall, many people with higher interest rates may be better off if they refinance their home mortgage. Refinancing is not always the correct move when interest rates fall. This is because many lending institutions charge stiff penalties to refinance.

The program shown in this section assumes that there are no penalties to be paid to refinance a mortgage.

The following program computes the new monthly payment, given the unpaid balance, the number of payments remaining to be paid, and the new interest rate available.

```
10 INPUT"ENTER UNPAID BALANCE";A
20 INPUT"ENTER # OF PAYMENTS REMAINING";N
30 INPUT"ENTER NEW INTEREST RATE AVAILABLE";I
40 I=I/1200
50 PM=A/((1-(1+I)↑-N)/I)
60 PM=INT(PM*100+.5)/100
70 PRINT"NEW MONTHLY PAYMENT: $";PM
```

SAMPLE PROBLEM

If the unpaid balance of your present loan is $61,215 and you have 325 payments remaining to be paid, what would your new monthly payment be if you were to refinance your loan at 9% interest?

```
ENTER UNPAID BALANCE ? 61215
ENTER # OF PAYMENTS REMAINING ? 325
ENTER NEW INTEREST RATE AVAILABLE ? 9
NEW MONTHLY PAYMENT: $ 503.51
```

MAXIMUM LOAN AMOUNT

When considering the purchase of a home or any other item that requires monthly payments, you should always determine whether you can afford the monthly payments. Many loans on the market today stipulate that the monthly payment for a home mortgage cannot exceed 25% of your gross monthly income.

The program demonstrated in this section assumes that the maximum payment also includes taxes and insurance. The 25% limitation can be altered for your situation by altering line 60 in the program. You can change the .25 to any value you desire as your maximum percent for the monthly payment.

The following program computes the maximum loan amount, given the gross monthly income, annual interest rate, time period of the loan in months, and the percentage of the loan for taxes and insurance.

```
10 INPUT"ENTER GROSS MONTHLY INCOME";IN
20 INPUT"ENTER ANNUAL INTEREST RATE AVAILABLE";I
```

```
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 INPUT"ENTER % OF LOAN PAYMENT FOR TAX & INS";T
50 I=I/1200
60 PM=(IN*.25)/(1+T/100)
70 A=PM*((1-(1+I)↑-N)/I)
80 A=INT(A*100+.5)/100
90 PRINT"MAXIMUM LOAN AMOUNT: $";A
```

SAMPLE PROBLEM

Your current income is $2,350 per month and the present interest rate available is 12%. What is the maximum loan amount you can afford on a 30-year (360-month) loan if 25% of your payment has to go for taxes and insurance? This loan requires that not more than 25% of your gross monthly income can be used for the mortgage payment.

```
ENTER GROSS MONTHLY INCOME ? 2350
ENTER ANNUAL INTEREST RATE AVAILABLE ? 12
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
ENTER % OF LOAN PAYMENT FOR TAX & INS ? 25
MAXIMUM LOAN AMOUNT: $ 45692.62
```

NUMBER OF PAYMENTS TO PAY OFF MORTGAGE

When assuming a loan, it is necessary to determine how many payments remain to pay off the loan. The program shown here allows you to perform this calculation. The program is also valuable if you wish to raise your monthly payment to pay off your loan at an earlier period.

The following program computes the number of monthly payments remaining to be paid on a loan. The inputs to this program are the loan balance, the monthly payment, and the interest rate.

```
10 INPUT"ENTER LOAN BALANCE";A
20 INPUT"ENTER MONTHLY PAYMENT";PM
30 INPUT"ENTER INTEREST RATE";I
40 I=I/1200
50 N=LOG(1-(A*I/PM))/-LOG(1+I)
60 N=INT(N*100+.5)/100
70 PRINT"NUMBER OF MONTHLY PAYMENTS: ";N
```

SAMPLE PROBLEM

If you decide to raise your monthly mortgage payment to $621 and the interest rate on your loan is 9%, how many monthly payments would remain to be paid if your current loan balance is $31,241?

```
ENTER LOAN BALANCE ? 31241
ENTER MONTHLY PAYMENT ? 621
ENTER INTEREST RATE ? 9
NUMBER OF MONTHLY PAYMENTS: 63.4
```

PRINCIPAL AND INTEREST FOR A SPECIFIC PAYMENT

When computing your taxes or doing other financial analysis on your home mortgage, it is sometimes valuable to determine how much of a specific payment is for principal and how much for interest. It's interesting to see that very little is paid toward the principal at the beginning of the loan period and a great amount is paid at the end of the loan.

The following program computes the amount toward the principal and the amount for interest of a specific payment. The inputs to the program are the initial loan amount, the interest rate, the monthly payment, and the number of the payment desired.

```
10 INPUT"ENTER INITIAL LOAN AMOUNT";A
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER MONTHLY PAYMENT";PM
40 INPUT"ENTER # OF PAYMENT DESIRED";P
50 I=I/1200:P=P-1
60 N=LOG(1-(A*I/PM))/-LOG(1+I)
70 X=PM*((1-(1+I)↑(-N+P))/I)
80 IP=X*I
90 PP=PM-IP
100 IP=INT(IP*100+.5)/100
110 PP=INT(PP*100+.5)/100
120 PRINT"AMOUNT TOWARD PRINCIPAL: $";PP
130 PRINT"AMOUNT FOR INTEREST: $";IP
```

SAMPLE PROBLEM

The initial amount of your loan is $48,500, with an interest rate of 11.5% and a monthly payment of $525. What portion of payment number 50 is for principal and what amount is for interest?

```
ENTER INITIAL LOAN AMOUNT ? 48500
ENTER INTEREST RATE ? 11.5
ENTER MONTHLY PAYMENT ? 525
ENTER # OF PAYMENT DESIRED ? 50
AMOUNT TOWARD PRINCIPAL: $ 96.08
AMOUNT FOR INTEREST: $ 428.92
```

INTEREST PAID FOR EACH YEAR OF MORTGAGE

The federal government allows you to deduct interest expenses on your home mortgage from your taxable income. This is usually a very sizable amount and one of the largest deductions available to you.

Your lending agency is required to send you a statement at the end of the year to inform you how much interest you have paid. For planning purposes, it sometimes helps to calculate this amount at an earlier time. This program will allow you to do so.

The following program computes the yearly interest and accumulated interest for each year of your loan. The inputs to the program are the loan amount, the interest rate, the time period of the loan in months, and the number of payments in the first year of the loan.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 INPUT"ENTER # OF PAYMENTS IN FIRST YEAR";FY
50 I=I/1200
60 PM=A/((1-(1+I)↑-N)/I)
70 PRINT"YEAR  YEARLY INT  ACCUMULATED INT"
80 PV=PM*((1-(1+I)↑-N)/I)
90 X=PM*((1-(1+I)↑(-N+FY))/I)
100 Z=FY*PM+X-PV
110 Z=INT(Z*100+.5)/100
120 Y=1
130 PRINT Y;TAB(6)Z;TAB(20)Z
140 TZ=Z
150 FY=FY+12
160 IF FY=>N THEN FY=N
170 X=PM*((1-(1+I)↑(-N+FY))/I)
180 Z=FY*PM+X-PV
190 Z=Z-TZ
200 Z=INT(Z*100+.5)/100
```

```
210 TZ=TZ+Z
220 Y=Y+1
230 PRINT Y;TAB(6)Z;TAB(20)TZ
240 IF FY=N THEN END
250 GOTO 150
```

SAMPLE PROBLEM

If you have a loan amount of $125,000, with an interest rate of 12.5% over a 20-year (240-month) period, what is the yearly interest if there are 8 payments made in the first year?

```
ENTER LOAN AMOUNT? 125000
ENTER INTEREST RATE? 12.5
ENTER TIME PERIOD OF LOAN (MONTHS) ? 240
ENTER # OF PAYMENTS IN FIRST YEAR ? 8
YEAR  YEARLY INT   ACCUMULATED INT
1     10381.50     10381.50
2     15411.17     25792.67
3     15195.20     40987.87
4     14950.65     55938.52
5     14673.71     70612.23
6     14360.09     84972.32
7     14004.94     98977.26
8     13602.78     112580.04
9     13147.36     125727.40
10    12631.63     138359.03
11    12047.61     150406.64
12    11386.26     161792.90
13    10637.33     172430.23
14    9789.24      182219.47
15    8828.85      191048.32
16    7741.27      198789.59
17    6509.70      205299.29
18    5115.03      210414.32
19    3535.70      213950.02
20    1747.23      215697.25
21    144.91       215842.16
```

MONTHLY PAYMENT REQUIRED WITH BALLOON PAYMENT

Some mortgages call for a very large final payment at the completion of the loan. This is known as a balloon payment. With the balloon payment the series of payments are not equal, so the standard program for computing the required monthly payment cannot be used.

The following program computes the monthly payment required, given the loan amount, the interest rate, the time period of the loan in months, and the balloon payment.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 INPUT"ENTER BALLOON PAYMENT";BP
50 I=I/1200
60 PV=BP*(1+I)↑-N
70 PV=A-PV
80 N=N-1
90 PM=PV/((1-(1+I)↑-N)/I)
100 PM=INT(PM*100+.5)/100
110 PRINT"MONTHLY PAYMENT REQUIRED: $";PM
```

SAMPLE PROBLEM

You purchased a home with a mortgage amount of $56,300 and an interest rate of 10.75%. What is the monthly payment required if it is a 20-year (240-month) loan with a balloon payment of $5,000?

```
ENTER LOAN AMOUNT ? 56300
ENTER INTEREST RATE ? 10.75
ENTER TIME PERIOD OF LOAN (MONTHS) ? 240
ENTER BALLOON PAYMENT ? 5000
MONTHLY PAYMENT REQUIRED: $ 566.28
```

MAXIMUM LOAN AMOUNT WITH BALLOON PAYMENT

When determining how to finance a purchase, in many cases it's a good idea to structure it with a balloon payment. This is especially useful if you expect to receive a future lump sum at the time the balloon payment is due.

This program allows you to alter the regular monthly payment and the balloon payment to see how it affects the maximum loan you can afford. If you expect the interest rate to be less at the end of the loan period, you can have quite a large balloon payment and then plan on refinancing when it is due.

The following program computes the maximum loan amount, given the regularly monthly payment, the interest rate, the time period of the loan in months, and the balloon payment.

```
10 INPUT"ENTER REGULAR MONTHLY PAYMENT";PM
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 INPUT"ENTER BALLOON PAYMENT";BP
50 I=I/1200
60 PV=BP*(1+I)↑-N
70 N=N-1
80 P=PM*((1-(1+I)↑-N)/I)
90 PV=PV+P
100 PV=INT(PV*100+.5)/100
110 PRINT"MAXIMUM LOAN AMOUNT: $";PV
```

SAMPLE PROBLEM

If you could afford a monthly payment of $970 with an interest rate of 11.25% over a 30-year (360-month) period, what is the maximum loan amount you could afford, with the final payment being a balloon payment of $20,000?

```
ENTER REGULAR MONTHLY PAYMENT ? 970
ENTER INTEREST RATE ? 11.25
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
ENTER BALLOON PAYMENT ? 20000
MAXIMUM LOAN AMOUNT: $ 100531.62
```

DETERMINING BALLOON PAYMENT

Numerous different methods can be set up to repay a loan, with a balloon payment being the final payment. This allows the borrower to set the loan amount and then pay off the remainder of the loan at any time with a balloon payment.

The following program computes the balloon payment required, given the loan amount, the interest rate, the time period of the loan in months, and the month the balloon payment is to be made.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
```

```
40 INPUT"ENTER MONTH BALLOON PAYMENT MADE";BM
50 I=I/1200
60 PM=A/((1-(1+I)↑-N)/I)
70 BP=PM*((1-(1+I)↑(-N+BM))/I)
80 BP=INT(BP*100+.5)/100
90 PRINT"BALLOON PAYMENT REQUIRED: $";BP
```

SAMPLE PROBLEM

If you had a loan of $83,143 with an interest rate of 12.75% over a 30-year (360-month) period, what would be the balloon payment required if you decided to pay off the loan on payment number 200?

```
ENTER LOAN AMOUNT ? 83143
ENTER INTEREST RATE ? 12.75
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
ENTER MONTH BALLOON PAYMENT MADE ? 200
BALLOON PAYMENT REQUIRED: $69361.53
```

PAYMENT REQUIRED WHEN COMBINING TWO MORTGAGES

When doing home improvement projects, or other activities that require you to borrow money, it may be better to finance it along with your home mortgage. This is not always true or feasible, but it's something that should be considered.

With the initial purchase of a home, it is usually easy to include such additions as a fence, a swimming pool, or other items with the home mortgage. If these things were financed individually, the lending agencies would not allow the payments to be spread over such a long period as a home mortgage.

The following program computes the combined monthly payments with two loans and the single payment with both items financed together. The inputs to the program are the loan amount, the interest rate, and the time period of the loan in months for both loans.

```
10 INPUT"ENTER INITIAL LOAN AMOUNT";A1
20 INPUT"ENTER INTEREST RATE";I1
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N1
40 INPUT"ENTER SECOND LOAN AMOUNT";A2
50 INPUT"ENTER INTEREST RATE";I2
60 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N2
```

```
70 I1=I1/1200
80 I2=I2/1200
90 P1=A1/((1-(1+I1)↑-N1)/I1)
100 P2=A2/((1-(1+I2)↑-N2)/I2)
110 A3=A1+A2
120 P3=A3/((1-(1+I1)↑-N1)/I1)
130 P1=P1+P2
140 P1=INT(P1*100+.5)/100
150 P3=INT(P3*100+.5)/100
160 PRINT"COMBINED PAYMENTS WITH TWO LOANS: $";P1
170 PRINT"SINGLE PAYMENT FINANCED TOGETHER: $";P3
```

SAMPLE PROBLEM

You are purchasing a home and want to put in a swimming pool. The loan that is available to finance the $95,000 required for your home is at 11.5% interest over a 30-year (360-month) period. Your $25,000 swimming pool can be financed either with your home mortgage or as a separate loan at 13% interest over a 10-year period. What would the monthly payments be with the two financing options?

```
ENTER INITIAL LOAN AMOUNT ? 95000
ENTER INTEREST RATE ? 11.5
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
ENTER SECOND LOAN AMOUNT ? 25000
ENTER INTEREST RATE ? 13
ENTER TIME PERIOD OF LOAN (MONTHS) ? 120
COMBINED PAYMENTS WITH TWO LOANS: $ 1314.05
SINGLE PAYMENT FINANCED TOGETHER: $ 1188.35
```

VALUATION OF RENTAL PROPERTY

When considering the purchase of a home, office building, or shopping center that you intend to rent, you should always determine its value. The program shown here demonstrates a method for doing this.

This valuation is based on a constant flow of income from the rental property, with consideration of the residual value of the property at the end of its lease life.

The following program computes the present valuation of rental property, given the monthly rental income, the leasing life of the property in years, the value of the property at the end of its lease life, and the interest rate.

```
10 INPUT"ENTER MONTHLY RENTAL INCOME";PM
20 INPUT"ENTER LEASING LIFE (YEARS)";N
30 INPUT"ENTER VALUE AT END OF LEASE LIFE";FV
40 INPUT"ENTER INTEREST RATE";I
50 I2=I/1200
60 I=I/100
70 N2=N*12
80 PV=PM*((1-(1+I2)↑-N2)/I2)
90 P2=FV*(1+I)↑-N
100 PV=PV+P2
110 PV=INT(PV*100+.5)/100
120 PRINT"PRESENT VALUATION OF PROPERTY: $";PV
```

SAMPLE PROBLEM

You are considering purchasing a small office building that can be rented for $5,000 per month. The expected leasing life of this building is 25 years, at which time it will be worth around $100,000. If the present interest rate is 13%, what is the value of this rental property?

```
ENTER MONTHLY RENTAL INCOME ? 5000
ENTER LEASING LIFE (YEARS) ? 25
ENTER VALUE AT END OF LEASE LIFE ? 100000
ENTER INTEREST RATE ? 13
PRESENT VALUATION OF PROPERTY: $ 448037.34
```

LEASE AMOUNT FOR RENTAL PROPERTY

Purchasing real estate for leasing purposes can be a very profitable endeavor if the rental lease amount provides a profitable return. The program demonstrated in this chapter determines the minimum lease amount to charge for rental property.

It should be noted that with this program, if the appreciation rate of the value of the property is greater than the interest rate, the rental lease amount will be negative. This is because this calculation only determines the minimum amount to charge to break even.

The following program computes the minimum rental lease amount, given the current value of the property, the leasing life in years, the property value at the end of the lease life, and the interest rate.

```
10 INPUT"ENTER CURRENT VALUE OF PROPERTY";PV
20 INPUT"ENTER LEASING LIFE (YEARS)";N
30 INPUT"ENTER VALUE AT END OF LEASE LIFE";FV
40 INPUT"ENTER INTEREST RATE";I
50 12=I/1200
60 I=I/100
70 N2=N*12
80 P2=FV*(1+I)↑-N
90 PM=(PV-P2)/((1-(1+I2)↑-N2)/I2)
100 PM=INT(PM*100+.5)/100
110 PRINT"RENTAL LEASE AMOUNT: $";PM
```

SAMPLE PROBLEM

A house that you have recently purchased cost you $61,300. If you expect to rent this house for 20 years, at which time it will be worth $50,000, what rental amount should you charge if the current interest rate is 12%?

```
ENTER CURRENT VALUE OF PROPERTY ? 61300
ENTER LEASING LIFE (YEARS) ? 20
ENTER VALUE AT END OF LEASE LIFE ? 50000
ENTER INTEREST RATE ? 12
RENTAL LEASE AMOUNT: $ 617.89
```

CHAPTER

5

Loan Analysis

When you take out a loan, you are using someone else's money and then repaying them, over the life of the loan, the amount you borrowed (the principal) and the interest on that money. The interest on a loan can be calculated in several different ways. The federal government regulations require that lenders quote or advertise interest rates as *A*nnual *P*ercentage *R*ates (APR). This helps with overall comparison of lending institutions and offers general protection for the consumer.

In this chapter we shall examine add-on interest rates and how to convert from add-on to APR and from APR to add-on. This is very useful when comparing these two different types of loans.

When considering a loan, the percent of the down payment is a very important factor. Also included in this chapter is a program showing the effects of different amounts of down payments on the monthly loan payment and on the total interest paid.

Another factor when analyzing loans is the taxation consideration. The interest portion of the monthly payment is tax deductible. The amount of interest and principal paid for each monthly payment is shown in the section on loan amortization.

One of the most popular aspects of loan analysis is the determination of the payment required for a specific loan. The standard equation for this calculation is as follows:

$$PM = AMT \div \frac{1 - (1 + I)^{-N}}{I}$$

where: PM = *monthly payment required*
AMT = *loan amount*

I = *periodic interest rate*
N = *number of monthly payments*

This equation can be easily restructured to solve for the loan amount, the interest rate, or the time period required. These calculations are shown with the programs in this chapter.

AMOUNT TO BORROW AND AMOUNT OF DOWN PAYMENT

When considering a purchase where a loan is involved, you first have to determine the monthly payment you can afford before making the purchase. You should make the decision by calculating the present interest rate to determine the loan you can afford and the down payment required.

The following program computes the down payment required and the loan amount, given the purchase amount, the monthly payment you can afford, the APR finance charge, and the time period of the loan in months.

```
10 INPUT"ENTER PURCHASE AMOUNT";A
20 INPUT"ENTER MONTHLY PAYMENT YOU CAN AFFORD";PM
30 INPUT"ENTER FINANCE CHARGE APR%";I
40 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
50 I=I/1200
60 PV=PM*((1-(1+I)↑-N)/I)
70 PV=INT(PV*100+.5)/100
80 X=A-PV
90 X=INT(X*100+.5)/100
100 PRINT"DOWN PAYMENT AMOUNT: $";X
110 PRINT"LOAN AMOUNT: $";PV
```

SAMPLE PROBLEM

If you are considering purchasing a new car for $9,800 and you can afford a monthly payment of $190, what loan and down payment amounts are required for a 48-month loan with 16.25% APR financing?

```
ENTER PURCHASE AMOUNT ? 9800
ENTER MONTHLY PAYMENT YOU CAN AFFORD ? 190
ENTER FINANCE CHARGE APR% ? 16.25
ENTER TIME PERIOD OF LOAN (MONTHS) ? 48
DOWN PAYMENT AMOUNT: $ 3125.95
LOAN AMOUNT: $ 6674.05
```

AMOUNT OF PAYMENT

In certain situations when analyzing a purchase that requires a loan, the payment amount must be calculated. This is one of the most used calculations when doing a loan analysis because it tells you the bottom line—what a purchase will cost you on a monthly basis.

The following program computes the monthly payment amount, given the loan amount, the APR finance charge, and the time period of the loan in months.

```
10 INPUT"ENTER PURCHASE AMOUNT";A
20 INPUT"ENTER FINANCE CHARGE APR%";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 I=I/1200
50 PM=A/((1-(1+I)↑-N)/I)
60 PM=INT (PM*100+.5)/100
70 PRINT"MONTHLY PAYMENT: $";PM
```

SAMPLE PROBLEM

If you are purchasing some furniture for $2,500 on a 36-month loan with 15% APR financing, what would your monthly payment be?

```
ENTER PURCHASE AMOUNT ? 2500
ENTER FINANCE CHARGE APR% ? 15
ENTER TIME PERIOD OF LOAN (MONTHS) ? 36
MONTHLY PAYMENT: $ 86.66
```

APR PERCENT INTEREST

In some situations when considering a loan, the purchase amount, the payment amount, and time period are predefined and you must determine what annual percentage interest rate is required to allow for this specified payment.

The following program computes the annual percentage rate (APR) required, given the loan amount, the payment amount, and the time period of the loan in months. If an invalid set of numbers is entered (for instance, if the monthly payment multiplied by the number of monthly payments is less than the loan amount), a message stating so is displayed.

```
10 INPUT"ENTER PURCHASE AMOUNT";A
20 INPUT"ENTER DOWN PAYMENT AMOUNT";DP
30 INPUT"ENTER PAYMENT AMOUNT";PM
40 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
50 A=A-DP
60 IF PM*N<A THEN PRINT"ERROR WITH INPUTS":END
70 X=A/PM
80 S=.001:BG=.0008
90 FOR K=BG TO .0833 STEP S
100 Z=(1-(1+K)↑-N)/K
110 IF Z>X THEN NEXT K
120 IF K>=.0833 THEN PRINT"RATE: 100+ %":END
130 IF S<>.001 THEN 150
140 S=.0001:BG=K-.001:GOTO 90
150 K=K*1200
160 K=INT(K*100+.5)/100
170 PRINT"INTEREST RATE % REQUIRED: ";K
```

SAMPLE PROBLEM

The house you are interested in is selling for $120,000. To purchase this home you must make a down payment of $30,000 and 360 monthly payments of $950. What interest rate is required to do this?

```
ENTER PURCHASE AMOUNT ? 120000
ENTER DOWN PAYMENT AMOUNT ? 30000
ENTER PAYMENT AMOUNT ? 950
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
INTEREST RATE % REQUIRED: 12.36
```

TIME REQUIRED TO PAY OFF LOAN

When considering a loan, sometimes the number of payments required to pay off the loan can be altered to fit financial commitments. This can be analyzed when the loan amount, payment amount, and interest rate are available.

The following program computes the number of monthly payments required given the loan amount, monthly payment amount, and the APR finance charge.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER MONTHLY PAYMENT AMOUNT";PM
```

```
30 INPUT"ENTER FINANCE CHARGE APR%";I
40 I=I/1200
50 N=LOG(1-(A*I/PM))/-LOG(1+I)
60 N=INT(N*100+.5)/100
70 PRINT"NUMBER OF MONTHLY PAYMENTS: ";N
```

SAMPLE PROBLEM

If you borrow $1,500 to buy a new television and make a monthly payment of $75, how many monthly payments must you make if the APR interest rate is 18%?

```
ENTER LOAN AMOUNT ? 1500
ENTER MONTHLY PAYMENT AMOUNT ? 75
ENTER FINANCE CHARGE APR% ? 18
NUMBER OF MONTHLY PAYMENTS: 23.96
```

INTEREST PAID

When making monthly payments on a loan, each payment is comprised of a certain portion to repay the loan and a certain portion to pay interest on the borrowed money. When doing a loan analysis, it is sometimes valuable to determine the amount of interest paid over the life of the loan.

The following program computes the accumulated interest paid over the life of a loan, given the loan amount, the APR finance charge, and the time period of the loan in months.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER FINANCE CHARGE APR%";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 I=I/1200
50 PM=A/((1-(1+I)↑-N)/I)
60 X=PM*N
70 X=X-A
80 X=INT(X*100+.5)/100
90 PRINT"ACCUMULATED INTEREST PAID: $";X
```

SAMPLE PROBLEM

If you borrow $20,000 over a time period of 10 years (120 months) and the APR finance charge is 12%, what is the accumulated interest paid over the life of the loan?

```
ENTER LOAN AMOUNT ? 20000
ENTER FINANCE CHARGE APR% ? 12
ENTER TIME PERIOD OF LOAN (MONTHS) ? 120
ACCUMULATED INTEREST PAID: $ 14433.03
```

COST OF TIME PAYMENTS

With certain sales promotions, a product is offered with a price tag of a monthly payment over a period of time. When considering this type of purchase, it's a good idea to calculate the initial purchase price (the loan amount) of the product for comparison to similar products priced in the standard way.

The following program computes the loan amount, given the monthly payment amount, the APR finance charge, and the number of monthly payments.

```
10 INPUT"ENTER MONTHLY PAYMENT AMOUNT";PM
20 INPUT"ENTER FINANCE CHARGE APR%";I
30 INPUT"ENTER NUMBER OF MONTHLY PAYMENTS";N
40 I=I/1200
50 PV=PM*((1-(1+I)↑-N)/I)
60 PV=INT(PV*100+.5)/100
70 PRINT"LOAN AMOUNT: $";PV
```

SAMPLE PROBLEM

If a car salesman wants to sell you a car with a monthly payment of $230 over 48 months and the APR finance charge is 18%, what is the loan amount for this purchase?

```
ENTER MONTHLY PAYMENT AMOUNT ? 230
ENTER FINANCE CHARGE APR% ? 18
ENTER NUMBER OF MONTHLY PAYMENTS ? 48
LOAN AMOUNT: $7829.79
```

ADD-ON INTEREST RATE TO APR

In certain situations when considering a loan, the lender may quote you an add-on interest rate. When comparing this loan to other types, it would be helpful to convert this rate to the standard APR rate.

The following program computes the equivalent APR interest rate, given the add-on rate, the loan amount, and the time period of the loan in months.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER ADD-ON FINANCE CHARGE %";FC
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 TP=A+(FC/100)*(N/12)*A
50 PM=TP/N
60 IF PM*N<A THEN PRINT"ERROR WITH INPUTS":END
70 X=A/PM
80 S=.001:BG=.0008
90 FOR K=BG TO .0833 STEP S
100 Z=(1-(1+K)↑-N)/K
110 IF Z>X THEN NEXT K
120 IF K>=.0833 THEN PRINT"RATE: 100+ %";END
130 IF S<>.001 THEN 150
140 S=.0001:BG=K-.001:GOTO 90
150 K=K*1200
160 K=INT(K*100+.5)/100
170 PRINT"EQUIVALENT APR% RATE: ";K
```

SAMPLE PROBLEM

If you were offered a loan of $2,000 over 24 months with an add-on interest rate of 8%, what is the equivalent APR rate?

```
ENTER LOAN AMOUNT ? 2000
ENTER ADD-ON FINANCE CHARGE % ? 8
ENTER TIME PERIOD OF LOAN (MONTHS) ? 24
EQUIVALENT APR% RATE: 14.76
```

APR TO ADD-ON INTEREST

When a lending institution quotes an APR interest rate, it may be valuable to convert this to an add-on rate for comparison purposes. It may also be useful to do this if the lender quotes the loan both ways to see if they are correct.

The following program computes the equivalent add-on interest rate, given the APR rate, the loan amount, and the time period of the loan in months.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER FINANCE CHARGE APR%";I
```

```
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 I=I/1200
50 PM=A/((1-(1+I)↑-N)/I)
60 X=(PM*N)-A
70 X=((X*100)/(N/12))/A
80 X=INT(X*100+.5)/100
90 PRINT"EQUIVALENT ADD-ON RATE: ";X
```

SAMPLE PROBLEM

If you were offered a loan of $5,000 over 36 months with an APR finance charge of 18%, what is the equivalent add-on rate?

```
ENTER LOAN AMOUNT ? 5000
ENTER FINANCE CHARGE APR% ? 18
ENTER TIME PERIOD OF LOAN (MONTHS) ? 36
EQUIVALENT ADD-ON RATE: 10.05
```

SAVINGS BENEFIT FROM DOWN PAYMENT

When considering a loan, it is always important to determine what different amounts of down payment will do to the loan. The amount of interest paid changes greatly with different percentages of the down payment.

The following program computes a schedule for the total interest paid and the payment required with a range of down payment percentages from 0 to 90. The loan amount, the APR finance charge, and the time period of the loan in months are to be specified.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER FINANCE CHARGE APR%";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 I=I/1200:A2=A
50 PRINT
60 PRINT"% DOWN  PAYMENT  INTEREST PAID"
70 PM=A/((1-(1+I)↑-N)/I)
80 X=PM*N
90 X=X-A
100 X=INT(X*100+.5)/100
110 PM=INT(PM*100+.5)/100
120 PRINT Z,PM,X
130 Z=Z+10
140 IF Z=100 THEN END
```

```
150 A=A2*(1-(Z*.01))
160 GOTO 70
```

SAMPLE PROBLEM

If you are going to purchase a house for $67,500 with an APR finance charge of 11.5% over a period of 360 months, what will be the different monthly payments and the total interest paid for various down payment amounts?

```
ENTER LOAN AMOUNT ? 67500
ENTER FINANCE CHARGE APR% ? 11.5
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
% DOWN   PAYMENT   INTEREST PAID
0        668.45    173140.82
10       601.60    155826.74
20       534.76    138512.65
30       467.91    121198.57
40       401.07    103884.49
50       334.22    86570.41
60       267.38    69256.33
70       200.53    51942.25
80       133.69    34628.16
90       66.84     17314.08
```

REDUCTION IN FINANCE CHARGE WITH RULE OF 78

Whenever a loan is paid off early, a portion of the finance charge is returned to the borrower. Many retailers use the rule of 78, which returns a portion of the finance charge according to the sum of the months' digits remaining with the loan.

The following program computes the total finance charge and the refunded portion of the finance charge, given the loan amount, the APR percentage finance charge, the time period of the loan in months, and the number of months remaining on the loan when it is paid off.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER FINANCE CHARGE APR%";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 INPUT"ENTER # OF MONTHS REMAINING WHEN PAID";RM
50 I=I/100
60 FC=A*I*(N/12)
```

```
70 CP=(RM*(RM+1))/2
80 TP=(N*(N+1))/2
90 R=(CP*FC)/TP
100 R=INT(R*100+.5)/100
110 FC=INT(FC*100+.5)/100
120 PRINT"TOTAL FINANCE CHARGE: $";FC
130 PRINT"REFUNDED FINANCE CHARGE: $";R
```

SAMPLE PROBLEM

If you purchased a refrigerator for $950 with an APR finance charge of 18% on a 36-month note, what would be the total finance charge if you paid it off with 15 months remaining on the loan?

```
ENTER LOAN AMOUNT ? 950
ENTER FINANCE CHARGE APR% ? 18
ENTER TIME PERIOD OF LOAN (MONTHS) ? 36
ENTER # OF MONTHS REMAINING WHEN PAID ? 15
TOTAL FINANCE CHARGE: $ 513
REFUNDED FINANCE CHARGE: $ 92.43
```

AMORTIZING A LOAN

With each monthly payment of a loan, a portion of the payment is for the interest charge and a portion is to repay the loan. This breakdown is very important with regard to the computation of federal income taxes, since interest is a deductible item.

The following program computes an amortization schedule of a loan, given the loan amount, the APR finance charge, and the time period of the loan in months.

It should be noted that this program pauses when the screen is full and then continues when the RETURN key is pressed.

```
10 INPUT"ENTER LOAN AMOUNT";A
20 INPUT"ENTER FINANCE CHARGE APR%";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 J=I/1200:CT=1
50 P=A*(J/(1-1/(1+J)↑N))
60 P=INT(P*100+.5)/100
70 PRINT"PAYMENT AMOUNT: $";P
80 PRINT"PYMT#  INTEREST PD  PRINCIPAL PD"
```

```
90 R=A*J:Q=P-R:A=A-Q:S=S+1
100 R=INT(R*100+.5)/100
110 Q=INT(Q*100+.5)/100
120 PRINT S,R,Q
130 IF S=N THEN END
140 CT=CT+1
150 IF CT<23 GOTO 90
160 CT=0
170 INPUT"PRESS <RETURN> TO CONTINUE";Q
180 GOTO 80
```

SAMPLE PROBLEM

If you borrow $80,000 at 12% APR financing for 360 months, what is the portion of each payment toward interest and principal?

```
ENTER LOAN AMOUNT ? 80000
ENTER FINANCE CHARGE APR% ? 12
ENTER TIME PERIOD OF LOAN (MONTHS) ? 360
PAYMENT AMOUNT: $ 822.89
PYMT#  INTEREST PD  PRINCIPAL PD
1      800.00       22.89
2      799.77       23.12
3      799.54       23.35
4      799.31       23.58
5      799.07       23.82
6      798.83       24.06
7      798.59       24.30
8      798.35       24.54
9      798.10       24.79
10     797.86       25.03
11     797.61       25.29
12     797.35       25.54
13     797.10       25.79
14     796.84       26.05
15     796.58       26.31
16     796.32       26.57
17     796.05       26.84
18     795.78       27.11
19     795.51       27.38
20     795.24       27.65
21     794.96       27.93
22     794.68       28.21
PRESS <RETURN> TO CONTINUE
```

CHAPTER 6

Savings

Savings and loan companies and other types of financial institutions pay individuals a certain amount for allowing them to use their money for a specific time period. The payment is based on a specific interest rate, and the sum is compounded at whatever periodic time span is set up by the agency.

There are numerous kinds of savings accounts. Each type has its own advantages and disadvantages as to flexibility, interest rates, withdrawal options, methods for computing interest on deposits and withdrawals, and compounding periods. The methods shown with the programs in this chapter are very generalized and standard applications.

When planning your financial options with regard to savings, you can make a single deposit and then allow it to grow, or you can adjust your budget such that you make consistent periodic deposits. Both of these savings plans are analyzed in this chapter.

One of the most used calculations with regard to savings applications is the determination of the future value of a savings account. The standard equation for this calculation is as follows:

$$FV = DEP \times \left[1 + \frac{I}{M}\right]^{[N \times M]}$$

where: FV = *future value of savings account*
DEP = *initial deposit in savings account*
I = *interest rate*
M = *number of compound periods per year*
N = *number of years invested*

This equation can be easily rearranged to solve for other variables. This is demonstrated with programs in this chapter.

FUTURE VALUE OF A SAVINGS DEPOSIT

If you deposit a certain amount of money in a savings account, it will earn a specific interest rate and will grow to a future value in a specified number of years.

The following program computes the future value of a single savings deposit, given the initial deposit, the interest rate, the number of compound periods per year, and the number of years it is left in the savings account.

```
10 INPUT"ENTER INITIAL AMOUNT OF SAVINGS";A
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER # OF COMPOUND PERIODS/YEAR";M
40 INPUT"ENTER NUMBER OF YEARS INVESTED";N
50 K=K/100
60 FV=A*(1+K/M)↑(N*M)
70 FV=INT(FV*100+.5)/100
80 PRINT"FUTURE VALUE OF SAVINGS: $";FV
```

SAMPLE PROBLEM

If you deposit $12,000 in a savings account that earns 7.25% interest compounded daily, what would be the value of this account in 5 years?

```
ENTER INITIAL AMOUNT OF SAVINGS ? 12000
ENTER INTEREST RATE ? 7.25
ENTER # OF COMPOUND PERIODS/YEAR ? 365
ENTER NUMBER OF YEARS INVESTED ? 5
FUTURE VALUE OF SAVINGS: $ 17242.41
```

DEPOSIT REQUIRED FOR SPECIFIED FUTURE AMOUNT

In planning your future financial needs, it is sometimes advantageous to determine how much you should put in a savings account so that it would grow to a specified amount to cover a particular need.

The following program computes the deposit required to grow to a certain amount, given the desired future amount, the interest rate, the number of compound periods per year, and the number of years invested.

```
10 INPUT"ENTER DESIRED FUTURE AMOUNT";FV
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER # OF COMPOUND PERIODS/YEAR";M
40 INPUT"ENTER NUMBER OF YEARS INVESTED";N
50 K=K/100
60 PV=FV/((1+K/M)↑(N*M))
70 PV=INT(PV*100+.5)/100
80 PRINT"DEPOSIT REQUIRED: $";PV
```

SAMPLE PROBLEM

If you need $50,000 to send your child to college in 10 years, and your savings account pays 6.5% interest compounded monthly, how much should you deposit to reach this amount?

```
ENTER DESIRED FUTURE AMOUNT ? 50000
ENTER INTEREST RATE ? 6.5
ENTER # OF COMPOUND PERIODS/YEAR ? 12
ENTER NUMBER OF YEARS INVESTED ? 10
DEPOSIT REQUIRED: $26148.11
```

TIME REQUIRED FOR SPECIFIED FUTURE AMOUNT

If you want to analyze future financial needs and you know you have a certain amount to invest in savings at the present time, it can be valuable to determine how long it will take to reach your goal with this current sum.

The following program computes the number of years required to reach a specified amount, given the desired future amount, the interest rate, the number of compound periods per year, and the initial savings deposit.

```
10 INPUT"ENTER INITIAL DEPOSIT";PV
20 INPUT"ENTER DESIRED FUTURE AMOUNT";FV
30 INPUT"ENTER # OF COMPOUND PERIODS/YEAR";M
40 INPUT"ENTER INTEREST RATE";K
50 K=K/100
60 N=LOG(FV/PV)/(M*LOG(1+K/M))
70 N=INT(N*100+.5)/100
80 PRINT"NUMBER OF YEARS REQUIRED: ";N
```

SAMPLE PROBLEM

If you deposited $10,000 in a savings account paying 7% interest compounded quarterly, how long would it take for this account to reach $20,000?

```
ENTER INITIAL DEPOSIT ? 10000
ENTER DESIRED FUTURE AMOUNT ? 20000
ENTER # OF COMPOUND PERIODS/YEAR ? 4
ENTER INTEREST RATE ? 7
NUMBER OF YEARS REQUIRED: 9.99
```

INTEREST REQUIRED FOR SPECIFIED FUTURE AMOUNT

Another factor to consider when analyzing future financial needs is the interest that savings institutions are paying. It is sometimes important to know what interest rate is required for your savings to grow to a future desired amount.

The following program computes the interest rate required for your savings to grow to a specific amount, given the initial savings amount, the final amount desired, the number of years invested, and the number of compound periods per year.

```
10 INPUT"ENTER INITIAL SAVINGS AMOUNT";P
20 INPUT"ENTER FINAL AMOUNT DESIRED";A
30 INPUT"ENTER NUMBER OF YEARS INVESTED";T
40 INPUT"ENTER # OF COMPOUND PERIODS/YEAR";C
50 I=((A/P)↑(1/(C*T))-1)*C
60 I=I*100
70 I=INT(I*100+.5)/100
80 PRINT"REQUIRED RATE OF RETURN:";I
```

SAMPLE PROBLEM

If you currently have $5,000 in your savings account that compounds monthly, what interest rate would you have to get for this account to reach $10,000 in 10 years?

```
ENTER INITIAL SAVINGS AMOUNT ? 5000
ENTER FINAL AMOUNT DESIRED ? 10000
ENTER NUMBER OF YEARS INVESTED ? 10
ENTER # OF COMPOUND PERIODS/YEAR ? 12
REQUIRED RATE OF RETURN: 6.95
```

FINAL AMOUNT FROM EQUAL MONTHLY DEPOSITS

Some individuals find that the easiest way to save money is to consistently make specific periodic deposits. Some employers assist in this savings effort by offering payroll deductions to a savings account.

The following program computes the final amount of a series of monthly deposits, given the monthly deposit amount, the interest rate, and the number of months that deposits are made.

```
10 INPUT"ENTER DEPOSIT AMOUNT";PM
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF MONTHS";N
40 K=K/1200
50 FV=PM*(((1+K)↑N-1)/K)
60 FV=INT(FV*100+.5)/100
70 PRINT"FINAL AMOUNT OF SAVINGS: $";FV
```

SAMPLE PROBLEM

If you make a monthly deposit of $150 for 10 years (120 months) and receive an interest rate of 6%, what would be the final amount in the account at the end of this period?

```
ENTER DEPOSIT AMOUNT ? 150
ENTER INTEREST RATE ? 6
ENTER NUMBER OF MONTHS ? 120
FINAL AMOUNT OF SAVINGS: $ 24581.90
```

PERIODIC DEPOSIT REQUIRED FOR SPECIFIED FUTURE AMOUNT

If you save by making periodic deposits, it can be to your advantage to determine the necessary deposit to reach a desired future amount. If this calculation is done, you'll have no surprises and you'll be better able to plan your financial commitments.

The following program computes the monthly deposit necessary to reach a desired amount in the future, given the desired future amount, the interest rate, and the number of months that deposits are made.

```
10 INPUT"ENTER DESIRED FUTURE AMOUNT";FV
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF MONTHS SAVED";N
```

```
40 K=K/1200
50 PM=FV/(((1+K)↑N-1)/K)
60 PM=INT(PM*100+.5)/100
70 PRINT"MONTHLY DEPOSIT REQUIRED: $";PM
```

SAMPLE PROBLEM

If you need $30,000 in 10 years (120 months) to send your children to college, and the interest rate is 6.5%, how much must you deposit each month to reach this future amount?

```
ENTER DESIRED FUTURE AMOUNT ? 30000
ENTER INTEREST RATE ? 6.5
ENTER NUMBER OF MONTHS SAVED ? 120
MONTHLY DEPOSIT REQUIRED: $ 178.14
```

TIME REQUIRED FOR SPECIFIED FUTURE AMOUNT WITH MONTHLY DEPOSITS

When saving by making monthly deposits, it's sometimes useful to determine how long it will take to reach a desired future amount. It's particularly helpful when this calculation can be used to tell you when, in the future, you can make a large purchase or pay off a debt.

The following program computes the number of monthly deposits required to reach a desired future amount, given the future amount desired, the interest rate, and the monthly deposit.

```
10 INPUT"ENTER DESIRED FUTURE AMOUNT";FV
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER MONTHLY DEPOSIT";PM
40 K=K/1200
50 N=LOG((K*FV)/PM+1)/LOG(1+K)
60 N=INT(N*100+.5)/100
70 PRINT"NUMBER OF MONTHLY DEPOSITS: ";N
```

SAMPLE PROBLEM

If you deposit $75 a month into an account that earns 7.5% interest, how many monthly deposits would be required for this account to reach $5,000?

```
ENTER DESIRED FUTURE AMOUNT ? 5000
ENTER INTEREST RATE ? 7.5
ENTER MONTHLY DEPOSIT ? 75
NUMBER OF MONTHLY DEPOSITS: 55.9
```

FUTURE VALUE WITH BEGINNING BALANCE AND MONTHLY DEPOSITS

In certain instances when you start a savings program, you may already have an initial balance in your account. If this is the case, it's a good idea to determine the future value of the account when monthly deposits are made for a specific period of time.

The following program computes the future value of a savings account, given the beginning balance, the monthly deposit, the interest rate, and the number of monthly deposits that are made.

```
10 INPUT"ENTER INITIAL BALANCE";A
20 INPUT"ENTER DEPOSIT AMOUNT";PM
30 INPUT"ENTER INTEREST RATE";K
40 INPUT"ENTER NUMBER OF MONTHS";N
50 K2=K/100
60 K=K/1200
70 FV=PM*(((1+K)↑N-1)/K)
80 F=A*(1+K2)↑(N/12)
90 FV=FV+F
100 FV=INT(FV*100+.5)/100
110 PRINT"FINAL AMOUNT OF SAVINGS: $";FV
```

SAMPLE PROBLEM

If you have a savings account with a beginning balance of $3,000 and you make monthly deposits of $100 for 3 years (36 months) with an interest rate of 5.25%, how much money will be in the account at the end of this period?

```
ENTER INITIAL BALANCE ? 3000
ENTER DEPOSIT AMOUNT ? 100
ENTER INTEREST RATE ? 5.25
ENTER NUMBER OF MONTHS ? 36
FINAL AMOUNT OF SAVINGS: $ 7387.54
```

PAYMENT REQUIRED WITH BEGINNING BALANCE AND SPECIFIED FUTURE VALUE

In some instances in your financial planning, it is advantageous to calculate the payment required to reach a desired future amount when you start with a beginning balance.

The following program computes the payment required to reach a desired future amount, given the beginning balance, the desired future amount, the interest rate, and number of monthly payments made.

```
10 INPUT"ENTER INITIAL BALANCE";A
20 INPUT"ENTER DESIRED FUTURE BALANCE";FV
30 INPUT"ENTER INTEREST RATE";K
40 INPUT"ENTER NUMBER OF MONTHS";N
50 K2=K/100
60 K=K/1200
70 F=A*(1+K2)↑(N/12)
80 FV=FV-F
90 PM=FV/(((1+K)↑N-1)/K)
100 PM=INT(PM*100+.5)/100
110 PRINT"MONTHLY PAYMENT REQUIRED: $";PM
```

SAMPLE PROBLEM

If you have a savings account with an initial balance of $2,000 and want to have $10,000 in this account in 3 years (36 months), what should be your monthly deposit if the interest rate is 6.5%?

```
ENTER INITIAL BALANCE ? 2000
ENTER DESIRED FUTURE BALANCE ? 10000
ENTER INTEREST RATE ? 6.5
ENTER NUMBER OF MONTHS ? 36
MONTHLY PAYMENT REQUIRED: $ 191.36
```

CHAPTER

7

Lease Analysis

A lease is a means by which an individual or company can acquire the use of a physical item for a stated period of time. For some individuals and most companies, a lease represents an important source of financing. The lessee has 100% use of the leased item for the specified period of time, but the lessor retains title to it.

At the end of the lease period, the leased item may have some value left, for which it could then be sold. This is referred to as its residual value. Many types of machinery and other equipment will have no residual value at the end of their lease period; other items, such as real estate, may appreciate in value.

The lease payment represents a consistent cash flow stream to the lessor, such as an annuity. Since payments are normally made at the beginning of each month, it could be considered an annuity due.

The equation for determining the present value of a lease without a residual value is as follows:

$$PV = \left[PMT \times (1 + I)\right] \times \left[\frac{1 - (1 + I)^{-N}}{I}\right]$$

where: PV = *present value of the lease*
I = *interest rate*
PMT = *monthly lease payment*
N = *time period of the lease in months*

This equation can be easily rearranged to solve for other variables. This is demonstrated with programs in this chapter.

PRESENT VALUE OF LEASE WITHOUT RESIDUAL VALUE

When you consider leasing your house, a piece of machinery, or other item, it can be useful to ascertain the present value of the lease. This can be calculated by determining the present value of the cash flow stream. This section demonstrates a leasing situation without a residual value.

The following program computes the present value of a lease, given the monthly lease payment, the annual interest rate, and the time period of the lease in months.

```
10 INPUT"ENTER MONTHLY LEASE PAYMENT";PM
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER TIME PERIOD OF LEASE (MONTHS)";N
40 I=I/1200
50 PV=(PM*(1+I))*((1-(1+I)↑-N)/I)
60 PV=INT(PV*100+.5)/100
70 PRINT"PRESENT VALUE OF LEASE: $";PV
```

SAMPLE PROBLEM

If you lease a piece of machinery for $650 per month for a 3-year (36-month) period and the annual interest rate is 6%, what is the present value of the lease?

```
ENTER MONTHLY LEASE PAYMENT ? 650
ENTER INTEREST RATE? 6
ENTER TIME PERIOD OF LEASE (MONTHS) ? 36
PRESENT VALUE OF LEASE: $ 21472.99
```

PAYMENT AMOUNT OF LEASE WITHOUT RESIDUAL VALUE

Sometimes when you consider leasing an item, you may want to figure out what payment is necessary to produce a certain present value of the payments. This section demonstrates a leasing situation without a residual value.

The following program computes the payment amount of a lease, given the present value of the lease, the minimum annual return, and the time period of the lease in months.

```
10 INPUT"ENTER PRESENT VALUE OF LEASE";PV
20 INPUT"ENTER MINIMUM ANNUAL RETURN";I
30 INPUT"ENTER TIME PERIOD OF LEASE (MONTHS)";N
40 I=I/1200
```

```
50 PM=PV/((1+I)*((1-(1+I)↑-N)/I))
60 PM=INT(PM*100+.5)/100
70 PRINT"MONTHLY PAYMENT REQUIRED: $";PM
```

SAMPLE PROBLEM

If you have a lease that has a present value of $10,000 over a 12-month period and you require an 8% return, what must be the monthly lease payment?

```
ENTER PRESENT VALUE OF LEASE ? 10000
ENTER MINIMUM ANNUAL RETURN ? 8
ENTER TIME PERIOD OF LEASE (MONTHS) ? 12
MONTHLY PAYMENT REQUIRED: $ 864.12
```

INTEREST RATE ON LEASE WITHOUT RESIDUAL VALUE

Another item to consider when leasing is the percent return required on the lease, which allows you to determine how good an investment the lease is. This is demonstrated with a leasing situation without a residual value.

The following program computes the annual interest rate required, given the purchase price of the item, the monthly lease payment, and the time period of the lease in months.

```
10 INPUT"ENTER PURCHASE PRICE";PV
20 INPUT"ENTER MONTHLY LEASE PAYMENT";PM
30 INPUT"ENTER TIME PERIOD OF LEASE (MONTHS)";N
40 X=PV/PM
50 S=.001:BG=.0008
60 FOR I=BG TO .0833 STEP S
70 Z=(1+I)*((1-(1+I)↑-N)/I)
80 IF Z>X THEN NEXT I
90 IF S<>.001 THEN 110
100 S=.0001:BG=I-.001:GOTO 60
110 I=I*1200
120 I=INT(I*100+.5)/100
130 PRINT"INTEREST RATE % REQUIRED: ";I
```

SAMPLE PROBLEM

If you are considering leasing a computer that has a purchase price of $50,000, and it can be leased for $2,300 a month over a 2-year

(24-month) period, what is the interest rate charged if this item has no residual value at the end of this period?

```
ENTER PURCHASE PRICE ? 50000
ENTER MONTHLY LEASE PAYMENT ? 2300
ENTER TIME PERIOD OF LEASE (MONTHS) ? 24
INTEREST RATE % REQUIRED: 10.68
```

NUMBER OF PAYMENTS ON LEASE WITHOUT RESIDUAL VALUE

It is sometimes valuable, when analyzing a lease situation, to determine the number of payments required to cover the cost of the item and to allow a specific return. This section looks at such a situation with no residual value at the end of the lease.

The following program computes the number of payments on a lease, given the purchase price, the monthly lease payment, and the annual interest rate.

```
10 INPUT"ENTER PURCHASE PRICE";PV
20 INPUT"ENTER MONTHLY LEASE PAYMENT";PM
30 INPUT"ENTER INTEREST RATE";I
40 I=I/1200
50 N=-LOG(1-(I*PV)/(PM*(1+I)))/LOG(1+I)
60 N=INT(N*100+.5)/100
70 PRINT"NUMBER OF MONTHLY PAYMENTS REQUIRED: ";N
```

SAMPLE PROBLEM

If you lease an item that has a purchase price of $7,500 and make monthly lease payments of $225 with an interest rate of 13%, how many monthly payments must be made?

```
ENTER PURCHASE PRICE ? 7500
ENTER MONTHLY LEASE PAYMENT ? 225
ENTER INTEREST RATE ? 13
NUMBER OF MONTHLY PAYMENTS REQUIRED: 41.02
```

PRESENT VALUE OF LEASE WITH RESIDUAL VALUE

When you consider leasing an item, it's important to determine the present value of the cash flow stream. If the item has a value at the end of the lease period, it must be calculated into the cash flow stream.

The following program computes the present value of a lease, given the residual value, the monthly lease payment, the interest rate, and the time period of the lease in months.

```
10 INPUT"ENTER RESIDUAL VALUE";RV
20 INPUT"ENTER MONTHLY LEASE PAYMENT";PM
30 INPUT"ENTER INTEREST RATE";I
40 INPUT"ENTER TIME PERIOD OF LEASE (MONTHS)";N
50 I2=I/100
60 I=I/1200
70 Y=N/12
80 V=RV*(1+I2)↑-Y
90 PV=(PM*(1+I))*((1-(1+I)↑-N)/I)
100 PV=PV+V
110 PV=INT(PV*100+.5)/100
120 PRINT"PRESENT VALUE OF LEASE: $";PV
```

SAMPLE PROBLEM

If you lease an apartment for $500 per month for 2 years (24 months) and the interest rate is 8%, what is the present value of the lease, if at the end of the lease period the apartment is worth $30,000?

```
ENTER RESIDUAL VALUE ? 30000
ENTER MONTHLY LEASE PAYMENT ? 500
ENTER INTEREST RATE ? 8
ENTER TIME PERIOD OF LEASE (MONTHS) ? 24
PRESENT VALUE OF LEASE: $ 36849.14
```

PAYMENT AMOUNT OF LEASE WITH RESIDUAL VALUE

A lease provides a cash flow to the lessor of an item. If the present value of this cash flow stream is predefined, the monthly lease payment to allow this amount can be determined. This is demonstrated with the leased item having a residual value at the end of the lease period.

The following program computes the payment amount of a lease, given the current value, the residual value, the desired rate of return, and the time period of the lease in months.

```
10 INPUT"ENTER CURRENT VALUE";PV
20 INPUT"ENTER RESIDUAL VALUE";RV
30 INPUT"ENTER DESIRED RATE OF RETURN";I
```

```
40 INPUT"ENTER TIME PERIOD OF LEASE (MONTHS)";N
50 I2=I/100
60 I=I/1200
70 Y=N/12
80 V=RV*(1+I2)↑-Y
90 NP=PV-V
100 PM=NP/((1+I)*((1-(1+I)↑-N)/I))
110 PM=INT(PM*100+.5)/100
120 PRINT"MONTHLY PAYMENT REQUIRED: $";PM
```

SAMPLE PROBLEM

If you lease a computer that has a current value of $150,000 for a 3-year (36-month) period with a required rate of return of 18%, what is the required monthly lease payment if the machine is worth $70,000 at the end of this period?

```
ENTER CURRENT VALUE ? 150000
ENTER RESIDUAL VALUE ? 70000
ENTER DESIRED RATE OF RETURN ? 18
ENTER TIME PERIOD OF LEASE (MONTHS) ? 36
MONTHLY PAYMENT REQUIRED: $3825.24
```

INTEREST RATE ON LEASE WITH RESIDUAL VALUE

When considering the profitability of a lease, you must determine the rate of return yielded from the cash flow stream. The residual value of the item must be approximated to determine this rate of return.

The following program computes the interest rate required, given the purchase price, the residual value, the monthly lease payment, and the time period of the lease in months.

```
10 INPUT"ENTER PURCHASE PRICE";PV
20 INPUT"ENTER RESIDUAL VALUE";RV
30 INPUT"ENTER MONTHLY LEASE PAYMENT";PM
40 INPUT"ENTER TIME PERIOD OF LEASE (MONTHS)";N
50 S=.001:BG=.0008
60 FOR I=BG TO .0833 STEP S
70 I2=I*12
80 X=(PV-(RV*(1+I2)↑-(N/12)))/PM
90 Z=(1+I)*((1-(1+I)↑-N)/I)
100 IF Z>X THEN NEXT I
```

```
110 IF S<>.001 THEN 130
120 S=.0001:BG=I-.001:GOTO 60
130 I=I*1200
140 I=INT(I*100+.5)/100
150 PRINT"INTEREST RATE % REQUIRED: ";I
```

SAMPLE PROBLEM

If you lease a $3,000 item for 2 years (24 months) at $100 per month, what is the interest rate required if it has a value of $1,000 at the end of the lease period?

```
ENTER PURCHASE PRICE ? 3000
ENTER RESIDUAL VALUE ? 1000
ENTER MONTHLY LEASE PAYMENT ? 100
ENTER TIME PERIOD OF LEASE (MONTHS) ? 24
INTEREST RATE % REQUIRED: 10.44
```

NUMBER OF PAYMENTS ON LEASE WITH RESIDUAL VALUE

When you consider leasing an item, the most important aspect is the rate of return it will provide. In certain situations you may want to determine the number of monthly lease payments to provide this return. This is demonstrated with a leasing situation where the leased item has a residual value at the end of the lease period.

The following program computes the number of payments of a lease, given the purchase price, the residual value, the monthly lease payment, and the interest rate.

```
10 INPUT"ENTER PURCHASE PRICE";PV
20 INPUT"ENTER RESIDUAL VALUE";RV
30 INPUT"ENTER MONTHLY LEASE PAYMENT";PM
40 INPUT"ENTER INTEREST RATE";I
50 I2=I/100
60 I=I/1200
70 FOR N=1 TO 1000
80 X=RV*(1+I2)↑-(N/12)
90 Z=PM*(1+I)*((1-(1+I)↑-N)/I)
100 IF X+Z => PV THEN 120
110 NEXT N
120 PRINT"NUMBER OF MONTHLY PAYMENTS REQUIRED: ";N
```

SAMPLE PROBLEM

If you have an item that you could sell for $10,000 but decide to lease it for a monthly payment of $250 with an interest rate of 8%, how many monthly payments will be required if the item will have a residual value of $1,000 at the end of the lease period?

```
ENTER PURCHASE PRICE ? 10000
ENTER RESIDUAL VALUE ? 1000
ENTER MONTHLY LEASE PAYMENT ? 250
ENTER INTEREST RATE ? 8
NUMBER OF MONTHLY PAYMENTS REQUIRED: 43
```

PRESENT VALUE OF LEASE AFTER EACH PAYMENT

When considering different leasing situations, it is sometimes useful to determine the present value of the lease at various stages of the lease life. This is valuable when determining penalties for breaking the lease at different time periods during the lease and also when selling the lease to another party. This is demonstrated for a lease without a residual value.

The following program computes the present value of a lease with every payment, given the monthly lease payments, the interest rate, and the time period of the lease in months.

```
10 INPUT"ENTER MONTHLY LEASE PAYMENT";PM
20 INPUT"ENTER INTEREST RATE";I
30 INPUT"ENTER TIME PERIOD OF LEASE (MONTHS)";N
40 I=I/1200
50 PRINT"PAYMENT#  PRESENT VALUE"
60 FOR X=N TO 0 STEP -1
70 PV=(PM*(1+I))*((1-(1+I)↑-X)/I)
80 PV=INT(PV*100+.5)/100
90 PRINT N-X,PV
100 NEXT X
```

SAMPLE PROBLEM

If you lease a piece of machinery, paying $1,200 a month for 12 months, and the interest rate is 13%, what is the present value of the lease after each payment?

```
ENTER MONTHLY LEASE PAYMENT ? 1200
ENTER INTEREST RATE ? 13
```

```
ENTER TIME PERIOD OF LEASE (MONTHS) ? 12
PAYMENT#    PRESENT VALUE
0           13580.80
1           12514.92
2           11437.50
3           10348.41
4           9247.52
5           8134.70
6           7009.82
7           5872.76
8           4723.39
9           3561.56
10          2387.14
11          1200.00
12          0
```

TO LEASE OR TO BUY

In many business dealings, people have to make a decision as to whether to lease or to buy an item. This decision is based on such factors as down payment required, expected life of an item, current interest rates, and the purchase price. The program presented with this section assumes that the entire amount of the item will be financed. It should also be noted that the tax liability is not considered with this program. The interest paid to finance a loan is a tax deductible item.

The following program computes the present values of the purchase and of the lease of a specified item.

```
10 INPUT"ENTER PURCHASE PRICE";PV
20 INPUT"ENTER % FINANCE CHARGE";I
30 INPUT"ENTER TIME PERIOD OF LOAN (MONTHS)";N
40 INPUT"ENTER RESIDUAL VALUE";RV
50 INPUT"ENTER MONTHLY LEASE PAYMENT";LP
60 INPUT"ENTER MONTHLY MAINTENANCE COST";C
70 INPUT"ENTER INTEREST RATE YOU COULD EARN";IR
80 INPUT"ENTER COMPOUND PERIODS/YEAR";CP
90 I=I/1200
100 I2=IR/100
110 IR=IR/1200
120 PM=PV/((1-(1+I)↑-N)/I)
130 PM=PM+C
140 PV=PM*((1-(1+IR)↑-N)/IR)
150 P2=RV/((1+I2/CP)↑((N/12)*CP))
```

```
160 PV=PV-P2
170 PV=INT(PV*100+.5)/100
180 PRINT"PRESENT VALUE OF PURCHASE: $";PV
190 PV=(LP*(1+IR))*((1-(1+IR)↑-N)/IR)
200 PV=INT(PV*100+.5)/100
210 PRINT"PRESENT VALUE OF LEASE: $";PV
```

SAMPLE PROBLEM

You are considering purchasing a $12,000 car that you could finance at a 13% annual percentage rate over a 3-year (36-month) period. The maintenance would be $35 a month and you could sell the automobile for $5,000 at the end of the 3-year period. This same vehicle could be leased for $225 a month and you would not be responsible for any of the maintenance. If you could save your money in a savings account that paid 6% interest compounded monthly, would it be better to lease or to purchase?

```
ENTER PURCHASE PRICE ? 12000
ENTER % FINANCE CHARGE ? 13
ENTER TIME PERIOD OF LOAN (MONTHS) ? 36
ENTER RESIDUAL VALUE ? 5000
ENTER MONTHLY LEASE PAYMENT ? 225
ENTER MONTHLY MAINTENANCE COST ? 35
ENTER INTEREST RATE YOU COULD EARN ? 6
ENTER COMPOUND PERIODS/YEAR ? 12
PRESENT VALUE OF PURCHASE: $10262.91
PRESENT VALUE OF LEASE: $7432.96
```

CHAPTER

8

Time Value of Money

When making any type of financial decision, one of the most important aspects to consider is the time value of money. You should never make any type of financial move that would not give you a return that could be obtained through a guaranteed type of investment (e.g., savings account or U. S. savings bond).

In this chapter we examine the present value and the future value of several different types of cash flows. These present value and future value calculations demonstrate that income that is expected soon is valued more highly than income that is expected far in the future.

It should be noted that the interest rate in all programs in this chapter should be entered as a percentage value and should not be entered in its decimal format.

FUTURE VALUE OF A SINGLE DEPOSIT

When money is invested in a savings account or some other type of fund that earns a constant interest rate, it will grow to a certain amount in a specified number of years.

The following program computes the future value of a single deposit, given the initial deposit, the interest rate, the number of compound periods per year, and the number of years it is invested. It should be noted that the interest rate is a percentage value and should not be entered in its decimal format.

```
10 INPUT"ENTER INITIAL VALUE";A
20 INPUT"ENTER INTEREST RATE";K
```

```
30 INPUT"ENTER # OF COMPOUND PERIODS/YEAR";M
40 INPUT"ENTER NUMBER OF YEARS INVESTED";N
50 K=K/100
60 FV=A*(1+K/M)↑(N*M)
70 FV=INT(FV*100+.5)/100
80 PRINT"FUTURE VALUE: $";FV
```

SAMPLE PROBLEM

If you invest $5,000 in a savings account that earns 6% interest compounded monthly, what would be the amount in this account in 10 years?

```
ENTER INITIAL VALUE ? 5000
ENTER INTEREST RATE ? 6
ENTER # OF COMPOUND PERIODS/YEAR ? 12
ENTER NUMBER OF YEARS INVESTED ? 10
FUTURE VALUE: $ 9096.98
```

PRESENT VALUE OF A FUTURE RECEIPT

The value at the present time of a specific amount of money that will yield a certain amount at a particular future date is known as the present value of the sum of money. In many business transactions, a specific amount of work is agreed to be done for payment at a later date. When this is the case, it is always important to compute the present value of this amount.

The following program computes the present value of a future receipt, given the future receipt, interest rate, and number of years until payment.

```
10 INPUT"ENTER FUTURE RECEIPT";FV
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF YEARS";N
40 K=K/100
50 PV=FV*(1+K)↑-N
60 PV=INT(PV*100+.5)/100
70 PRINT"PRESENT VALUE: $";PV
```

SAMPLE PROBLEM

If you are writing a program that will earn $10,000 for you upon completion and the present interest rate is 6%, what is the present value of this payment if it takes you 2 years to finish the project?

```
ENTER FUTURE RECEIPT ? 10000
ENTER INTEREST RATE ? 6
ENTER NUMBER OF YEARS ? 2
PRESENT VALUE: $ 8899.96
```

FUTURE VALUE OF AN ORDINARY ANNUITY

An annuity is a series of payments of a fixed amount for a specific period of time. If the payments occur at the end of each period, then it is called an ordinary annuity. Most annuities are of this type.

The following program computes the future value of an ordinary annuity, given the payment amount, the interest rate, and the number of years that the payments are made. With this program the payment period is monthly.

```
10 INPUT"ENTER PAYMENT AMOUNT";PM
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF MONTHS";N
40 K=K/1200
50 FV=PM*(((1+K)↑N-1)/K)
60 FV=INT(FV*100+.5)/100
70 PRINT"FUTURE VALUE: $";FV
```

SAMPLE PROBLEM

If you have an ordinary annuity in which you make a payment of $150 a month for 5 years (60 months) with an interest rate of 8%, what will be the resulting cash value?

```
ENTER PAYMENT AMOUNT ? 150
ENTER INTEREST RATE ? 8
ENTER NUMBER OF MONTHS ? 60
FUTURE VALUE: $ 11021.53
```

PRESENT VALUE OF AN ORDINARY ANNUITY

As previously defined, an annuity is a series of payments of a fixed amount for a specific period of time. With an ordinary annuity, the payments occur at the end of each period.

After a certain number of payments have been made into an annuity, it has a specific value. At the outset of the annuity, this specific value can be discounted back as a present valuation of the annuity.

The following program computes the present value of an ordinary annuity, given the payment amount, the interest rate, and the number of years that the payments are made. With this program the payments are monthly.

```
10 INPUT"ENTER PAYMENT AMOUNT";PM
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF MONTHS";N
40 K=K/1200
50 PV=PM*((1-(1+K)↑-N)/K)
60 PV=INT(PV*100+.5)/100
70 PRINT"PRESENT VALUE: $";PV
```

SAMPLE PROBLEM

If you have an ordinary annuity that pays you $1,000 a month for 3 years (36 months) and the interest rate is 6%, what is the present value of this annuity?

```
ENTER PAYMENT AMOUNT ? 1000
ENTER INTEREST RATE ? 6
ENTER NUMBER OF MONTHS ? 36
PRESENT VALUE: $32871.02
```

FUTURE VALUE OF AN ANNUITY DUE

As previously defined, an annuity is a series of payments of a fixed amount for a specific period of time. If the payments are made at the beginning of each period, it is an annuity due.

The following program computes the future value of an annuity due, given the payment amount, the interest rate, and the number of years that the payments are made. In this program the payment period is monthly.

```
10 INPUT"ENTER PAYMENT AMOUNT";PM
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF MONTHS";N
40 K=K/1200
50 FV=PM*(1+K)*(((1+K)↑N-1)/K)
60 FV=INT(FV*100+.5)/100
70 PRINT"FUTURE VALUE: $";FV
```

SAMPLE PROBLEM

If you have an annuity due for which you make a monthly payment of $250 for 10 years (120 months) with an interest rate of 7.5%, what will be the resulting cash value?

```
ENTER PAYMENT AMOUNT ? 250
ENTER INTEREST RATE ? 7.5
ENTER NUMBER OF MONTHS ? 120
FUTURE VALUE: $ 44760.61
```

PRESENT VALUE OF AN ANNUITY DUE

As previously defined, an annuity is a series of payments of a fixed amount for a specific period of time. In the case of an annuity due, the payments occur at the beginning of each period.

After a certain number of payments have been made into an annuity, it has a specific value. At the outset of the annuity, this specific value can be discounted back as a present valuation of the annuity.

The following program computes the present value of an annuity due, given the payment amount, the interest rate, and the number of years that the payments are made. With this program the payments are monthly.

```
10 INPUT"ENTER PAYMENT AMOUNT";PM
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF MONTHS";N
40 K=K/1200
50 PV=PM*(1+K)*((1-(1+K)↑-N)/K)
60 PV=INT(PV*100+.5)/100
70 PRINT"PRESENT VALUE: $";PV
```

SAMPLE PROBLEM

If you have an annuity due that yields a payment of $120 a month for 5 years (60 months) and the interest rate is 5.25%, what is the present value of the annuity?

```
ENTER PAYMENT AMOUNT ? 120
ENTER INTEREST RATE ? 5.25
ENTER NUMBER OF MONTHS ? 60
PRESENT VALUE: $6348.10
```

PRESENT VALUE OF A PERPETUITY

A perpetuity is an infinite series of payments of a fixed amount. These are not used much today but were very popular back in the 1800s to pay off huge government loans.

The following program computes the present value of a perpetuity, given the payment amount and the interest rate.

```
10 INPUT"ENTER PAYMENT AMOUNT";PM
20 INPUT"ENTER INTEREST RATE";K
30 K=K/100
40 PV=PM/K
50 PV=INT(PV*100+.5)/100
60 PRINT"PRESENT VALUE OF PERPETUITY: $";PV
```

SAMPLE PROBLEM

If you make perpetuity payments of $100 and the interest rate is 6%, what is its present value?

```
ENTER PAYMENT AMOUNT ? 100
ENTER INTEREST RATE ? 6
PRESENT VALUE OF PERPETUITY: $ 1666.67
```

CHAPTER

9

Stocks and Bonds Analysis

The ownership or equity of a corporation is held by its stockholders. The stockholders are the individuals or other firms who have bought a portion of the corporation by purchasing shares of stock in it. The incentive to purchase stock in a company is the belief that the firm will provide a greater return than the common stable investments such as savings accounts and certificates of deposits.

Some of the programs in this chapter allow you to determine the value of a stock and the rate of return on a stock. These are both calculated with a stock that pays constant dividends. Many corporations pay out constant dividends because that allows the investor to plan on the income and it gives the firm the appearance of being a stable and growing company.

Stock represents ownership in a corporation, and bonds represent the corporation's debt. A firm that needs to increase its working capital may choose to sell bonds instead of issuing more stock.

The current interest rate available in the market place will determine the selling price of the bonds. If the present interest rate is lower than the interest rate at the time the bonds were issued, the price of the bonds will be higher than face value. If the present interest rate is higher than the interest rate at the time the bonds were issued, the price of the bonds will be lower than face value.

VALUATION OF STOCK PAYING CONSTANT DIVIDENDS

When you consider purchasing a stock, you should first determine its theoretical value. This can be done by approximating what the stock will be selling for at a specific future date and incorporating it into the equation with the program presented here.

The following program computes the value of a stock, given the constant quarterly dividend, the time period to evaluate the stock (in years), the estimated stock price at the end of this period, and the desired rate of return.

```
10 INPUT"ENTER CONSTANT QUARTERLY DIVIDEND";D
20 INPUT"ENTER PERIOD TO EVALUATE STOCK (YEARS)";N
30 PRINT"ENTER ESTIMATED STOCK PRICE"
40 INPUT"AT END OF PERIOD";P
50 INPUT"ENTER DESIRED RATE OF RETURN";I
60 I=I/400
70 N=N*4
80 PV=P/(1+I)↑N
90 P2=D*(1+I)*((1-(1+I)↑-N)/I)
100 PV=PV+P2
110 PV=INT(PV*100+.5)/100
120 PRINT"PRESENT VALUATION OF STOCK: $";PV
```

SAMPLE PROBLEM

If you own stock that pays a constant quarterly dividend of $.75 and you expect an 18% return on this investment, what is the value of this stock if you expect it to be selling for $50 per share in 5 years?

```
ENTER CONSTANT QUARTERLY DIVIDEND ? .75
ENTER PERIOD TO EVALUATE STOCK (YEARS) ? 5
ENTER ESTIMATED STOCK PRICE
AT END OF PERIOD ? 50
ENTER DESIRED RATE OF RETURN ? 18
PRESENT VALUATION OF STOCK: $ 30.93
```

RATE OF RETURN ON STOCK PAYING CONSTANT DIVIDENDS

Investing in the stock market can be a very risky financial move compared to the stable financial returns that are available with savings accounts and certificates of deposits. With the latter two types of investments, the rate of return is well defined; with stocks, it is not quite so straightforward. The following program determines the rate of return on a stock. It uses an iterative process to determine the result, so it may take a few seconds to perform the calculation.

The following program computes the average rate of return on a stock. The inputs to the program are the price paid for the stock, the time lapsed since purchase (years), constant quarterly dividend, and the current stock price.

```
10 INPUT"ENTER PRICE PAID FOR STOCK";PD
20 INPUT"ENTER TIME LAPSED SINCE PURCHASE";N
30 INPUT"ENTER CONSTANT QUARTERLY DIVIDEND";D
40 INPUT"ENTER CURRENT STOCK PRICE";CP
50 N=N*4
60 S=.005:BG=.0025
70 FOR I=BG TO 2.5 STEP S
80 X1=CP/(1+I)↑N
90 X2=D*(1+I)*((1-(1+I)↑-N)/I)
100 IF PD>=(X1+X2) THEN 120
110 NEXT I
120 IF S=.0001 THEN 150
130 S=.0001:BG=I-.005
140 GOTO 70
150 I=I*400
160 I=INT(I*100+.5)/100
170 PRINT"AVERAGE RATE OF RETURN: ";I
```

SAMPLE PROBLEM

If you paid $24 a share for a stock that has had constant quarterly dividends of $.50 and the stock is selling for $31 a share 3 years later, what has been the average rate of return on the stock?

```
ENTER PRICE PAID FOR STOCK ? 24
ENTER TIME LAPSED SINCE PURCHASE ? 3
ENTER CONSTANT QUARTERLY DIVIDEND ? .50
ENTER CURRENT STOCK PRICE ? 31
AVERAGE RATE OF RETURN: 16.4
```

AFTER-TAX RETURN ON STOCK PAYING CONSTANT DIVIDENDS

It's sometimes quite difficult to determine the true return on a stock because of dividends paid, fluctuation of the stock price, commissions to buy and sell, and the taxation involved. The following program determines the rate of return with regard to taxation considerations. It uses an iterative process to determine the result, so it may take a few seconds to perform the calculation.

The following program computes the average rate of return on a stock paying constant dividends. The inputs to the program are the price paid for the stock, the time period between purchasing and selling, the constant quarterly dividend, the current stock price, the tax rate percentage on dividends, and the tax rate percentage on the gain.

```
10 INPUT"ENTER PRICE PAID FOR STOCK";PD
20 INPUT"ENTER TIME LAPSED SINCE PURCHASE";N
30 INPUT"ENTER CONSTANT QUARTERLY DIVIDEND";D
40 INPUT"ENTER CURRENT STOCK PRICE";CP
50 INPUT"ENTER TAX RATE % ON DIVIDENDS";TD
60 INPUT"ENTER TAX RATE % ON GAIN";TG
70 N=N*4
80 GN=CP-PD
90 GN=GN*(TG/100)
100 CP=CP-GN
110 D=D*(1-TD/100)
120 S=.005:BG=.0025
130 FOR I=BG TO 2.5 STEP S
140 X1=CP/(1+I)↑N
150 X2=D*(1+I)*((1-(1+I)↑-N)/I)
160 IF PD>=(X1+X2) THEN 180
170 NEXT I
180 IF S=.0001 THEN 210
190 S=.0001:BG=I-.005
200 GOTO 130
210 I=I*400
220 I=INT(I*100+.5)/100
230 PRINT"AVERAGE RATE OF RETURN: ";I
```

SAMPLE PROBLEM

If you purchased a stock for $42.50 that has had constant quarterly dividends of $.71, and the stock is selling for $64.25 4 years later, what has been the average rate of return on the stock with regard to tax considerations? (The tax rate on dividends is 40%, and the tax rate on the gain is 18%.)

```
ENTER PRICE PAID FOR STOCK ? 42.50
ENTER TIME LAPSED SINCE PURCHASE ? 4
ENTER CONSTANT QUARTERLY DIVIDEND ? .71
ENTER CURRENT STOCK PRICE ? 64.25
ENTER TAX RATE % ON DIVIDENDS ? 40
```

```
ENTER TAX RATE % ON GAIN ? 18
AVERAGE RATE OF RETURN: 12.4
```

VALUATION OF A GROWTH STOCK

Some stocks experience a supernormal growth period because of change in consumer demands or for other reasons. This supernormal growth rate must be for a specified period of time, after which the stock returns to its normal growth rate.

The following program computes the present value of a growth stock, given the current yearly dividend, the normal growth rate, the supernormal growth rate, the supernormal growth period in years, and the expected rate of return.

```
10 INPUT"ENTER CURRENT YEARLY DIVIDEND";D
20 INPUT"ENTER NORMAL GROWTH RATE";NG
30 INPUT"ENTER SUPERNORMAL GROWTH RATE";SG
40 INPUT"ENTER SUPERNORMAL GROWTH PERIOD";N
50 INPUT"ENTER EXPECTED RATE OF RETURN";K
60 NG=NG/100
70 SG=SG/100
80 K=K/100
90 X=((1+K)/(1+SG))-1
100 PV=((D*(1+NG)/(K-NG))*(1+X)↑-N)+(D*((1-(1+X)↑-N)/X))
110 PV=INT(PV*100+.5)/100
120 PRINT"PRESENT VALUE OF GROWTH STOCK: $";PV
```

SAMPLE PROBLEM

If you own a growth stock that experiences a supernormal growth rate of 20% over a 3-year period and then returns to its normal growth rate of 7%, what is the present value of the stock? (The current yearly dividend is $2.00 and you expect a 16% rate of return.)

```
ENTER CURRENT YEARLY DIVIDEND ? 2.00
ENTER NORMAL GROWTH RATE ? 7
ENTER SUPERNORMAL GROWTH RATE ? 20
ENTER SUPERNORMAL GROWTH PERIOD ? 3
ENTER EXPECTED RATE OF RETURN ? 16
PRESENT VALUE OF GROWTH STOCK: $ 32.75
```

VALUATION OF A BOND

A holder of a bond will usually receive $1,000 with the maturity of the bond. Some bonds are sold with different maturity values, but this is the most common. When considering the value of a bond, you must consider items other than just the maturity value of the bond. These items include the time value of money, the coupon payments, the interest rate on the bond, and the current interest rate available.

The following program computes the value of the bond each year until maturity. The inputs to the program are the amount of coupon interest paid yearly, the maturity value of the bond, the current return available, and the number of years until maturity.

```
10 INPUT"ENTER AMOUNT OF YEARLY COUPON PAYMENT";I
20 INPUT"ENTER MATURITY VALUE OF BOND";M
30 INPUT"ENTER CURRENT RETURN AVAILABLE";K
40 INPUT"ENTER # OF YEARS TO MATURITY";N
50 K=K/100
60 PRINT"YEAR  BOND VALUE"
70 FOR X=N TO 0 STEP -1
80 Y=I*((1-(1/(1+K)↑X))/K)
90 Z=M*(1/(1+K))↑X
100 V=Y+Z
110 V=INT(V*100+.5)/100
120 PRINT N-X;TAB(6)V
130 NEXT X
```

SAMPLE PROBLEM

If you purchase a $1,000 bond that has a $90 coupon payment per year and it matures in 8 years, what is the value of the bond each year until maturity if the present required return on the market is 12%?

```
ENTER AMOUNT OF YEARLY COUPON PAYMENT ? 90
ENTER MATURITY VALUE OF BOND ? 1000
ENTER CURRENT RETURN AVAILABLE? 12
ENTER # OF YEARS TO MATURITY ? 8
YEAR  BOND VALUE
0     850.97
1     863.09
2     876.66
3     891.86
4     908.88
```

```
5       927.95
6       949.30
7       973.21
8       1000.00
```

CURRENT YIELD OF A BOND

When you purchase a bond, the price you pay may be more or less than the maturity value, depending on the other available investment yields on the market. It's useful to know the current yield on a bond you buy. There are also other considerations for tax purposes, but these are not taken into account with this program.

The following program computes the current yield of a bond, given the purchase price of the bond, the yearly interest received, and the capital gain received.

```
10 INPUT"ENTER PURCHASE PRICE OF BOND";P
20 INPUT"ENTER YEARLY INTEREST RECEIVED";I
30 INPUT"ENTER CAPITAL GAIN RECEIVED";G
40 X=(I+G)/P
50 X=X*100
60 X=INT(X*100+.5)/100
70 PRINT"CURRENT YIELD OF BOND: ";X
```

SAMPLE PROBLEM

If you purchased a bond for $940.32 and then sold it one year later, what would be the current yield on this bond if you received an interest payment of $95 and a capital gain of $2.60?

```
ENTER PURCHASE PRICE OF BOND ? 940.32
ENTER YEARLY INTEREST RECEIVED ? 95
ENTER CAPITAL GAIN RECEIVED ? 2.60
CURRENT YIELD OF BOND: 10.38
```

YIELD TO MATURITY OF A BOND

When considering whether to purchase a bond, you should determine its yield to maturity. This is the interest rate discussed by bond traders when they talk about rates of return.

The following program computes the yield to maturity, given the price of the bond, maturity value of the bond, the number of years to maturity, and the interest rate of the bond.

```
10 INPUT"ENTER PRICE OF BOND";V
20 INPUT"ENTER MATURITY VALUE OF BOND";M
30 INPUT"ENTER # OF YEARS TO MATURITY";N
40 INPUT"ENTER INTEREST RATE OF BOND";I
50 I=M*I/100
60 YM=(I+(M-V)/N)/((M+V)/2)
70 YM=YM*100
80 YM=INT(YM*100+.5)/100
90 PRINT"YIELD TO MATURITY: ";YM
```

SAMPLE PROBLEM

If you could purchase a 10-year, 9% coupon, $1,000 maturity value bond at a price of $1,075, what would be the yield to maturity?

```
ENTER PRICE OF BOND ? 1075
ENTER MATURITY VALUE OF BOND ? 1000
ENTER # OF YEARS TO MATURITY ? 10
ENTER INTEREST RATE OF BOND ? 9
YIELD TO MATURITY: 7.95
```

AMORTIZING A BOND

When a company issues a bond offering, the entire bond issue can be amortized over the life of the bond. This amortization shows the interest expense and bond discount for each period.

The following program computes the interest expense and bond discount for each coupon period until the bond matures. The inputs to the program are the total value of the bond issue, the total selling price of the bond issue, the interest rate with semiannual coupon, the annual yield to maturity, and the number of years to maturity.

```
10 INPUT"ENTER TOTAL VALUE OF BONDS";TV
20 INPUT"ENTER TOTAL SELLING PRICE OF BONDS";TP
30 PRINT"ENTER INTEREST RATE WITH SEMIANNUAL"
40 INPUT"COUPON";I
50 INPUT"ENTER ANNUAL YIELD TO MATURITY";YM
60 INPUT"ENTER # OF YEARS TO MATURITY";N
```

```
70 I=(I/100)/2*TV
80 YM=(YM/100)/2
90 N=N*2
100 PRINT"PERIOD VALUE       INT EXPENSE DISCOUNT"
110 FOR X=1 TO N
120 BI=TP*YM
130 A=BI-I
140 TP=INT(TP*100+.5)/100
150 BI=INT(BI*100+.5)/100
160 A=INT(A*100+.5)/100
170 PRINT X;TAB(6)TP;TAB(18)BI;TAB(30)A
180 TP=TP+A
190 NEXT X
```

SAMPLE PROBLEM

If a company issues $300,000 worth of bonds for $283,000 and the bonds mature in 7 years, the interest rate is 7% with a semiannual coupon, and its yield to maturity is 8%, what is the amortized bonds' value, discount, and interest for each period?

```
ENTER TOTAL VALUE OF BONDS ? 300000
ENTER TOTAL SELLING PRICE OF BONDS ? 283000
ENTER INTEREST RATE WITH SEMIANNUAL
COUPON ? 7
ENTER ANNUAL YIELD TO MATURITY ? 8
ENTER # OF YEARS TO MATURITY ? 7
PERIOD  VALUE       INT EXPENSE  DISCOUNT
1       283000.00   11320.00     820.00
2       283820.00   11352.80     852.80
3       284672.80   11386.91     886.91
4       285559.71   11422.39     922.39
5       286482.10   11459.28     959.28
6       287441.38   11497.66     997.66
7       288439.04   11537.56     1037.56
8       289476.60   11579.06     1079.06
9       290555.66   11622.23     1122.23
10      291677.89   11667.12     1167.12
11      292845.01   11713.80     1213.80
12      294058.81   11762.35     1262.35
13      295321.16   11812.85     1312.85
14      296634.01   11865.36     1365.36
```

CHAPTER

10

Sinking Fund Analysis

When a firm has future plans to replace worn equipment, add additional facilities, or retire a bond issue, it often establishes a sinking fund into which it makes fixed periodic payments in order to reach a specific lump sum amount in the future. A sinking fund is very similar to an ordinary annuity. The major difference is that most sinking fund situations involve finding the payment required instead of the maturity value.

When computing a sinking fund problem the following information is needed:

1. the future amount desired;
2. the payment amount;
3. the annual interest rate;
4. the number of years or number of payments to be made.

The programs in this chapter solve for each of these variables, given the other three variables.

FUTURE VALUE

As defined in the introduction of this chapter, a sinking fund is a series of payments of a fixed amount over a specific period of time. When analyzing the return on a sinking fund, you must determine the future value of the payments over a certain time period.

The following program computes the future value of a sinking fund, given the payment amount, the annual interest rate, and the number of years that the payments are made. With this program, the payment periods are monthly.

```
10 INPUT"ENTER DEPOSIT AMOUNT";PM
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF YEARS";N
40 K=K/1200:N=N*12
50 FV=PM*(((1+K)↑N-1)/K)
60 FV=INT(FV*100+.5)/100
70 PRINT"FUTURE VALUE OF FUND: $";FV
```

SAMPLE PROBLEM

If a company paid $20,000 a month into a sinking fund for 5 years and the annual interest rate is 6%, what would be the future value of the fund?

```
ENTER DEPOSIT AMOUNT ? 20000
ENTER INTEREST RATE ? 6
ENTER NUMBER OF YEARS ? 5
FUTURE VALUE OF FUND: $ 1395400.60
```

PAYMENT AMOUNT

The most-used calculation with regard to sinking funds is that of the periodic payment required to attain a specific amount in the future. This is very important to businesses that need to pay off a large lump sum at a future date.

The following program computes the payment required, given the desired future amount, the annual interest rate, and the number of years that the payments are made. With this program the payment periods are monthly.

```
10 INPUT"ENTER FUTURE AMOUNT DESIRED";FV
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER NUMBER OF YEARS";N
40 K=K/1200:N=N*12
50 PM=(FV*K)/((1+K)↑N-1)
60 PM=INT(PM*100+.5)/100
70 PRINT"PAYMENT AMOUNT REQUIRED: $";PM
```

SAMPLE PROBLEM

What is the monthly payment required if $75,000 is needed in 8 years and the annual interest rate is 5.25%?

```
ENTER FUTURE AMOUNT DESIRED ? 75000
ENTER INTEREST RATE ? 5.25
ENTER NUMBER OF YEARS ? 8
PAYMENT AMOUNT REQUIRED: $ 630.32
```

INTEREST RATE

When working with a sinking fund, it is sometimes useful to determine what interest rate is required to obtain a specific amount in the future.

The following program computes the interest rate required to obtain a specific amount, given the monthly payment amount and the number of years that the payments are made. If the number of payments made, multiplied by the payment, is greater than the desired future amount, an "ERROR WITH INPUTS" message is displayed.

```
10 INPUT"ENTER FUTURE AMOUNT DESIRED";FV
20 INPUT"ENTER DEPOSIT AMOUNT";DP
30 INPUT"ENTER NUMBER OF YEARS";N
40 N=N*12
50 X=FV/DP
60 IF DP*N>FV THEN PRINT"ERROR WITH INPUTS":END
70 S=.001:BG=.0008
80 FOR K=BG TO .0833 STEP S
90 Z=((1+K)↑N-1)/K
100 IF Z<X THEN NEXT K
110 IF K>=.0833 THEN PRINT"RATE: 100+ %":END
120 IF S<>.001 THEN 140
130 S=.0001:BG=K-.001:GOTO 80
140 K=K*1200
150 K=INT(K*100+.5)/100
160 PRINT"INTEREST RATE % REQUIRED: ";K
```

SAMPLE PROBLEM

What annual interest rate percentage is required if $25,000 is needed in 2 years and a monthly deposit of $1,000 is made for this period?

```
ENTER FUTURE AMOUNT DESIRED ? 25000
ENTER DEPOSIT AMOUNT ? 1000
ENTER NUMBER OF YEARS ? 2
INTEREST RATE % REQUIRED: 4.32
```

NUMBER OF PAYMENTS

Sometimes an individual or a company budgets a certain amount for a savings account or other type of fund, which could eventually be used to pay off a future lump sum debt. If the budgeted monthly deposit is constant and the future debt is known, the time period to pay off this debt can be determined.

The following program computes the number of years required to reach a specific amount in a sinking fund, given the monthly deposit, the interest rate, and the future amount of the fund desired.

```
10 INPUT"ENTER FUTURE AMOUNT DESIRED";FV
20 INPUT"ENTER INTEREST RATE";K
30 INPUT"ENTER DEPOSIT AMOUNT";DP
40 K=K/1200
50 N=LOG((K*FV/DP)+1)/LOG(1+K)
60 N=N/12
70 N=INT(N*100+.5)/100
80 PRINT"NUMBER OF YEARS TO REACH AMOUNT: ";N
```

SAMPLE PROBLEM

If a company deposits $1,500 a month into a sinking fund and the annual interest rate is 7%, how many years would it take to have $100,000 in this fund?

```
ENTER FUTURE AMOUNT DESIRED ? 100000
ENTER INTEREST RATE ? 7
ENTER DEPOSIT AMOUNT ? 1500
NUMBER OF YEARS TO REACH AMOUNT: 4.71
```

CHAPTER

11

Forecasting Inventory Needs

In any business environment where physical items are sold, inventory control is always an important concern. You should have enough to meet consumer demand at all times, but not so much that you are tying up all your cash reserves that could be earning interest.

Forecasting inventory needs is a very hard task. No one can consistently predict the future correctly. The programs in this chapter predict future demand with regard to a constant growth rate. This is not always the case, but it is the most straightforward method to determine a future demand forecast.

This chapter also covers an analysis of cost-of-inventory needs and forecasting costs in comparison to demand. In almost all situations, the rise in product cost and consumer demand are not evenly correlated because of inflation.

INVENTORY NEEDS WITH CONSTANT INCREASE IN CONSUMER DEMAND

In any business environment, it is always important to forecast your inventory needs. If this can be done accurately you can control your costs in an efficient manner. In most inventory forecasting situations, demand is considered to be constant.

The following program computes the future demand for a specified number of years, given the present demand, yearly percent increase in demand, and the specified number of years to display.

```
10 INPUT"ENTER PRESENT DEMAND";PD
20 INPUT"ENTER YEARLY % INCREASE IN DEMAND";I
```

```
30 INPUT"ENTER # OF YEARS TO DISPLAY";Y
40 X2=PD
50 PRINT"YEAR #  FUTURE DEMAND"
60 FOR X=0 TO Y
70 PRINT X,X2
80 X2=X2*(1+I/100)
90 X2=INT(X2)
100 NEXT X
```

SAMPLE PROBLEM

If you are selling personal computers at your store and the present demand is 750 per year but you expect a 20% increase per year for the next 5 years, what will be the demand for these years?

```
ENTER PRESENT DEMAND ? 750
ENTER YEARLY % INCREASE IN DEMAND ? 20
ENTER # OF YEARS TO DISPLAY ? 5
YEAR #  FUTURE DEMAND
0       750
1       900
2       1080
3       1296
4       1555
5       1866
```

COST OF INVENTORY NEEDS

When forecasting consumer demand and inventory needs, you should also consider the increase in costs of this inventory. The increase in consumer demand rises at one rate, while the cost of the inventory rises at another rate. These should both be considered when doing forecasting and inventory analysis.

The following program computes the future demand and cost for a specified number of years, given the present demand, yearly percent increase in demand, cost per item, yearly percent increase in cost, and the number of years to display.

```
10 INPUT"ENTER PRESENT DEMAND";PD
20 INPUT"ENTER YEARLY % INCREASE IN DEMAND";I
30 INPUT"ENTER COST PER ITEM";C
40 INPUT"ENTER YEARLY % INCREASE IN COST";CI
50 INPUT"ENTER # OF YEARS TO DISPLAY";Y
60 X2=PD:C2=C
```

```
70 PRINT"YEAR #  FUTURE DEMAND  COST/ITEM"
80 FOR X=0 TO Y
90 PRINT X;TAB(13)X2;TAB(25)C2
100 X2=X2*(1+I/100)
110 X2=INT(X2)
120 C2=C2*(1+CI/100)
130 C2=INT(C2*100+.5)/100
140 NEXT X
```

SAMPLE PROBLEM

Let us say you are selling a product that costs you $5 per unit and you expect this cost to increase at a rate of 15% per year. If the demand for this product is 2,000 per year, with an expected increase of 20% per year for 3 years, what is the future demand and cost of this product?

```
ENTER PRESENT DEMAND ? 2000
ENTER YEARLY % INCREASE IN DEMAND ? 20
ENTER COST PER ITEM ? 5
ENTER YEARLY % INCREASE IN COST ? 15
ENTER # OF YEARS TO DISPLAY ? 3
YEAR #  FUTURE DEMAND  COST/ITEM
0       2000           5
1       2400           5.75
2       2880           6.61
3       3456           7.60
```

INVENTORY NEEDS CONSIDERING PRODUCT RETURNS

In certain industries, products are leased and then returned at a later time. When these products are returned, they are refurbished and then leased to someone else. In such situations, inventory control is very important regarding what is supplied as refurbished products and what as new products.

The following program computes future demand and calculates how many units are supplied as refurbished products and how many as new products. The items that are input for this computation are the present demand, the yearly percent increase in demand, the number of items returned, the yearly percent increase in returns, and the number of years to display.

```
10 INPUT"ENTER PRESENT DEMAND";PD
20 INPUT"ENTER YEARLY % INCREASE IN DEMAND";I
```

```
30 INPUT"ENTER # OF ITEMS RETURNED";R
40 INPUT"ENTER YEARLY % INCREASE IN RETURNS";RI
50 INPUT"ENTER # OF YEARS TO DISPLAY";Y
60 X2=PD:R2=R:N=X2-R2
70 PRINT"YEAR #  DEMAND  # REFURB  # NEW"
80 FOR X=0 TO Y
90 PRINT X;TAB(8)X2;TAB(20)R2;TAB(28)N
100 X2=X2*(1+I/100)
110 X2=INT(X2)
120 R2=R2*(1+RI/100)
130 R2=INT(R2)
140 N=X2-R2
150 NEXT X
```

SAMPLE PROBLEM

Say you own a typewriter rental agency and your present demand is 3,000 units per year but you expect it to increase at a rate of 20% per year. If 500 typewriters are returned to be refurbished each year and you expect this to increase by 12% each year, what is the future demand going to be, and how many units will be refurbished and how many will be new over the next 6-year period?

```
ENTER PRESENT DEMAND ? 3000
ENTER YEARLY % INCREASE IN DEMAND ? 20
ENTER # OF ITEMS RETURNED ? 500
ENTER YEARLY % INCREASE IN RETURNS ? 12
ENTER # OF YEARS TO DISPLAY ? 6
YEAR #  DEMAND  # REFURB  # NEW
0       3000    500       2500
1       3600    560       3040
2       4320    627       3693
3       5184    702       4482
4       6220    786       5434
5       7464    880       6584
6       8956    985       7971
```

FORECASTING TO PERCENTAGE INCREASE IN DEMAND

When forecasting inventory, the product demand is sometimes known for a 2-year period. From this information, the percentage increase can be deduced and a forecast can be calculated. The forecast presented here is based on constant growth.

The following program computes the percentage increase in demand and displays the future demand for the specified number of years. The items that are input for this computation are the demands for the first and second year and the specified number of years to display.

```
10 INPUT"ENTER DEMAND YEAR-1";D1
20 INPUT"ENTER DEMAND YEAR-2";D2
30 INPUT"ENTER # OF YEARS TO DISPLAY";Y
40 I=D2/D1
50 PRINT"PERCENTAGE INCREASE IN DEMAND: ";100*(I-1)
60 PRINT"YEAR #  FUTURE DEMAND"
70 Z=1
80 PRINT Z,D1
90 X2=D2
100 FOR X=2 TO Y
110 PRINT X,X2
120 X2=X2*I
130 X2=INT(X2)
140 NEXT X
```

SAMPLE PROBLEM

If you have a product of which you sold 5,000 units last year and you sell 7,200 units this year, what is the percentage increase in demand and what is the forecasted future demand with constant growth for 5 years?

```
ENTER DEMAND YEAR-1 ? 5000
ENTER DEMAND YEAR-2 ? 7200
ENTER # OF YEARS TO DISPLAY ? 5
PERCENTAGE INCREASE IN DEMAND: 44
YEAR #  FUTURE DEMAND
1       5000
2       7200
3       10368
4       14929
5       21497
```

CHAPTER

12

Payroll

There are numerous computer applications regarding the payroll functions of a company. These applications range from printing the checks to determining the salary increase of an employee.

Many companies quote employee salaries on an hourly, monthly, or annual basis. When comparing two jobs that quote salaries for different time periods, it's helpful to be able to convert these salaries from one time period to another. There are several sections in this chapter that demonstrate these conversions.

Many firms pay their employees a standard salary and also a commission relative to their performance. This is usually done in areas where sales are involved because it's a very definitive measure of productivity. A program regarding this subject is also shown in this chapter.

TAKE-HOME PAY

Almost everyone who receives a paycheck has a certain portion of it allocated for deductions such as taxes, insurance, and numerous other items. Once you total all the deductions, you can determine the percentage of the deductions, which is a percent of the total gross pay.

The following program computes your net take-home pay, given your gross income and the percentage of the deductions.

```
10 INPUT"ENTER GROSS INCOME";IN
20 INPUT"ENTER PERCENT OF DEDUCTIONS";D
30 X=IN*(1-D/100)
```

```
40 X=INT(X*100+.5)/100
50 PRINT"NET TAKE-HOME PAY: $";X
```

SAMPLE PROBLEM

If your gross weekly pay is $523.45 and the percentage for deductions is 23%, how much is your net take-home pay?

```
ENTER GROSS INCOME ? 523.45
ENTER PERCENT OF DEDUCTIONS ? 23
NET TAKE-HOME PAY: $ 403.06
```

OVERTIME EARNINGS

The federal government requires that firms engaged in interstate commerce pay one-and-a-half times the regular rate to any employee who has worked more than 40 hours a week. This is with the understanding that the first 40 hours are at the standard rate of pay.

The following program computes the overtime salary, the regular salary, and the total salary. The inputs to the program are the number of hours worked, the standard hourly rate, and the overtime hourly rate.

```
10 INPUT"ENTER NUMBER OF HOURS WORKED";H
20 INPUT"ENTER STANDARD HOURLY RATE";SR
30 INPUT"ENTER OVERTIME HOURLY RATE";ER
40 OH=H-40
50 X=OH*ER
60 X=INT(X*100+.5)/100
70 PRINT"OVERTIME SALARY: $";X
80 Y=40*SR
90 PRINT"REGULAR SALARY: $";Y
100 X=X+Y
110 PRINT"TOTAL SALARY: $";X
```

SAMPLE PROBLEM

If you worked a total of 53 hours in a week and your regular hourly salary is $7.00 per hour while your overtime rate is $10.50, what is your total weekly salary?

```
ENTER NUMBER OF HOURS WORKED ? 53
ENTER STANDARD HOURLY RATE ? 7.00
```

```
ENTER OVERTIME HOURLY RATE ? 10.50
OVERTIME SALARY: $ 136.50
REGULAR SALARY: $ 280
TOTAL SALARY: $ 416.50
```

PAYROLL DEDUCTIONS

In the preparation of a payroll check many items have to be deducted for various reasons. For each employee the percentage or amount is fixed for a period of time. The percentage and amounts are different for most employees.

The program shown in this section is a general example of a payroll deduction process. Most payroll systems have many more deduction types than shown here.

The following program computes the amount taken out for each specified deduction and the net take-home pay. Inputs to the program are the gross salary and the percentage for each indicated deduction.

```
10 INPUT"ENTER GROSS SALARY";S
20 INPUT"ENTER % FOR FEDERAL TAX";FT
30 INPUT"ENTER % FOR FICA TAX";FI
40 INPUT"ENTER % FOR STATE TAX";TX
50 INPUT"ENTER % FOR INSURANCE";IN
60 FT=INT(S*FT+.5)/100
70 FI=INT(S*FI+.5)/100
80 TX=INT(S*TX+.5)/100
90 IN=INT(S*IN+.5)/100
100 PRINT"FEDERAL TAX: $";FT
110 PRINT"FICA TAX: $";FI
120 PRINT"STATE TAX: $";TX
130 PRINT"INSURANCE: $";IN
140 X=S-FT-FI-TX-IN
150 PRINT"TAKE-HOME PAY: $";X
```

SAMPLE PROBLEM

Your gross monthly salary is $2,321.87. If 10% is deducted for federal taxes, 6% for FICA, 2% for state tax, and 1% for insurance, what amounts are taken out for each deduction and what is your net take-home pay?

```
ENTER GROSS SALARY ? 2321.87
ENTER % FOR FEDERAL TAX ? 10
```

```
ENTER % FOR FICA TAX ? 6
ENTER % FOR STATE TAX ? 2
ENTER % FOR INSURANCE ? 1
FEDERAL TAX: $ 232.19
FICA TAX: $ 139.31
STATE TAX: $ 46.44
INSURANCE: $ 23.22
TAKE-HOME PAY: $ 1880.71
```

ANNUAL AND MONTHLY SALARY RATES FROM HOURLY PAY

In most jobs, employees are paid on an hourly basis. It's occasionally useful to convert this hourly rate to an annual or monthly rate, most notably when comparing two jobs that quote pay rates for different time periods.

The following program computes the monthly and annual salaries, given the hourly rate. It assumes that 40 hours are worked each week.

```
10 INPUT"ENTER HOURLY PAY RATE";HP
20 AP=HP*2080
30 MP=AP/12
40 MP=INT(MP*100+.5)/100
50 PRINT"MONTHLY SALARY: $";MP
60 PRINT"ANNUAL SALARY: $";AP
```

SAMPLE PROBLEM

If you earn $8.72 per hour, what is your salary on monthly and annual bases?

```
ENTER HOURLY PAY RATE ? 8.72
MONTHLY SALARY: $ 1511.47
ANNUAL SALARY: $ 18137.60
```

HOURLY AND MONTHLY SALARY RATES FROM ANNUAL PAY

Some companies quote pay rates on an annual basis. When comparing this rate to those of other companies, it's sometimes necessary to convert this rate of pay to other time periods.

The following program computes the hourly and monthly salaries, given the annual rate. It assumes that 40 hours are worked each week.

```
10 INPUT"ENTER ANNUAL SALARY";AP
20 HP=INT((AP/2080)*100+.5)/100
30 MP=INT((AP/12)*100+.5)/100
40 PRINT"HOURLY SALARY: $";HP
50 PRINT"MONTHLY SALARY: $";MP
```

SAMPLE PROBLEM

If you earn $32,000 per year, what is your salary on hourly and monthly bases?

```
ENTER ANNUAL SALARY ? 32000
HOURLY SALARY: $ 15.38
MONTHLY SALARY: $ 2666.67
```

HOURLY AND ANNUAL SALARY RATES FROM MONTHLY PAY

Many government agencies pay their employees on a monthly basis. It is sometimes useful to know what this rate of pay is equivalent to at different time periods.

The following program computes the hourly and annual salaries, given the monthly rate. It assumes that 40 hours are worked each week.

```
10 INPUT"ENTER MONTHLY SALARY";MP
20 AP=MP*12
30 HP=INT((AP/2080)*100+.5)/100
40 PRINT"HOURLY SALARY: $";HP
50 PRINT"ANNUAL SALARY: $";AP
```

SAMPLE PROBLEM

If you earn $1894.56 monthly, what is your salary on hourly and annual bases?

```
ENTER MONTHLY SALARY ? 1894.56
HOURLY SALARY: $ 10.93
ANNUAL SALARY: $ 22734.72
```

ANNUAL SALARY WITH HOURLY RATE AND COMMISSIONS

In certain instances companies will pay an hourly wage and then subsidize this pay with commissions to motivate their staffs. This is usually done in the area of sales, where the commissions can be correlated to different sales amounts.

The following program computes the annual salary, given the hourly rate, the percentage commission on sales, and the average monthly sales volume. It assumes that 40 hours are worked each week.

```
10 INPUT"ENTER HOURLY PAY RATE";HP
20 INPUT"ENTER PERCENT COMMISSION ON SALES";CM
30 INPUT"ENTER AVERAGE MONTHLY SALES";MS
40 X=(HP*2080)+(MS*12)*(CM/100)
50 X=INT(X*100+.5)/100
60 PRINT"ANNUAL SALARY: $";X
```

SAMPLE PROBLEM

If you earn $6.00 per hour and get a 10% commission on sales, what is your annual salary if your average monthly sales are $20,000?

```
ENTER HOURLY PAY RATE ? 6.00
ENTER PERCENT COMMISSION ON SALES ? 10
ENTER AVERAGE MONTHLY SALES ? 20000
ANNUAL SALARY: $ 36480
```

SALARY INCREASES

When calculating or evaluating a salary increase, it is usually compared to your previous salary as a percentage. This percentage can be compared to previous raises you have received and also to the inflation rate.

The following program computes your new salary, given the initial salary and the percent increase.

```
10 INPUT"ENTER INITIAL SALARY";S
20 INPUT"ENTER PERCENT INCREASE";I
30 X=S*I/100
```

```
40 X=INT(X*100+.5)/100
50 PRINT"INCREASE AMOUNT: $";X
60 S=S+X
70 PRINT"NEW SALARY: $";S
```

SAMPLE PROBLEM

If your present salary is $24,500 annually and you receive an increase of 12%, what is your new salary?

```
ENTER INITIAL SALARY ? 24500
ENTER PERCENT INCREASE ? 12
INCREASE AMOUNT: $ 2940
NEW SALARY: $ 27440
```

SALARY INCREASE PERCENTAGE

When given a salary increase, it's often advantageous to know the percentage of the increase. This is very helpful in comparing it with your other salary increases, your peers, and the inflation rate. The percentage, not the amount, is usually the best way to evaluate a salary increase.

The following program computes the percentage increase, given the old salary and the new salary.

```
10 INPUT"ENTER OLD SALARY";OS
20 INPUT"ENTER NEW SALARY";NS
30 X=100*((NS-OS)/OS)
40 X=INT(X*100+.5)/100
50 PRINT"PERCENTAGE INCREASE: ";X
```

SAMPLE PROBLEM

If your old salary was $17,250 and your new salary is $19,400, what is the percentage increase?

```
ENTER OLD SALARY ? 17250
ENTER NEW SALARY ? 19400
PERCENTAGE INCREASE: 12.46
```

AVERAGE PERCENT INCREASE OVER SPECIFIC TIME PERIOD

Your salary increases can be evaluated over multiple years to determine the average percent increase. This average is very useful to compare against future raises. It should also be noted that the inflation rate could have a great effect on salary increases during certain periods of time.

The following program computes the average yearly salary increase. The inputs to the program are the previous salary, the current salary, and the number of years between the salaries. This is an iterative process, so it may take a few seconds to produce the result.

```
10 INPUT"ENTER PREVIOUS SALARY";PS
20 INPUT"ENTER CURRENT SALARY";CS
30 INPUT"ENTER # OF YEARS BETWEEN SALARIES";N
40 S=.005:BG=.0001
50 FOR I=BG TO 10 STEP S
60 X=CS*(1+I)↑-N
70 IF PS>=X THEN 90
80 NEXT I
90 IF S=.0001 THEN 120
100 S=.0001:BG=I-.005
110 GOTO 50
120 I=I*100
130 I=INT(I*100+.5)/100
140 PRINT"AVERAGE YEARLY SALARY INCREASE: ";I
```

SAMPLE PROBLEM

If you started at a company with a salary of $14,000 and your salary 5 years later is $35,000, what has been the average yearly increase in your salary?

```
ENTER PREVIOUS SALARY ? 14000
ENTER CURRENT SALARY ? 35000
ENTER # OF YEARS BETWEEN SALARIES ? 5
AVERAGE YEARLY SALARY INCREASE: 20.12
```

CHAPTER

13

Insurance

Insurance is purchased to minimize the cost of unforeseen losses. When you foresee a possibility of high risk, some type of insurance coverage may be warranted. There are numerous types of insurances available on the market today to protect your property, life, and health.

Loss protection is ensured through an insurance policy. This policy is a written contract that gives the owner a specified amount of coverage for a certain item. The policy also states the time period, deductible amount, special clauses, and other legal issues.

The insurance policy is purchased by making fixed periodic payments. These payments are called premiums. The insurance agent who sells the policy determines the premium required by looking it up in a table or using a special equation.

There are numerous types of insurance applications that can be computerized. A few are demonstrated in this chapter.

PREMIUM REQUIRED ON A POLICY

The cost of carrying an insurance policy is called the premium. This premium is calculated by using a specific rate for each $100 worth of insurance that is purchased. These rates vary according to criteria established by the insurance companies.

The following program computes the annual premium, given the amount of the insurance policy and the rate per $100 of insurance.

```
10 INPUT"ENTER INSURANCE POLICY AMOUNT";A
20 INPUT"ENTER ANNUAL RATE PER $100";R
```

```
30 X=(A/100)*R
40 X=INT(X*100+.5)/100
50 PRINT"ANNUAL PREMIUM: $";X
```

SAMPLE PROBLEM

If you purchased an insurance policy worth $50,000 and the annual rate is $.45 per $100, what is the premium required?

```
ENTER INSURANCE POLICY AMOUNT ? 50000
ENTER ANNUAL RATE PER $100 ? .45
ANNUAL PREMIUM: $ 225
```

INSURANCE COVERAGE WITH CO-INSURANCE CLAUSE

The co-insurance type of insurance coverage is used to discourage people from buying very small fire insurance policies. In this policy the insurance company states that they will pay the full fire loss only if the owner insures at least a certain percentage of the value of the property. If the property is not insured for this specific percent, the insurance company will pay only part of the loss and the owner must pay the remainder.

In the program demonstrated here, it assumes an 80% co-insurance clause. If the owner insures his property for at least 80% of the real value, the insurance company will pay the full amount of the loss, up to the amount that the policy was written for. If the property is insured for less than 80% of its real value, only a portion of the loss is paid by the insurance company. This percentage is defined as the amount of the insurance policy divided by the real value of the property.

The following program computes the amount the insurance company pays with the co-insurance clause. Inputs to the program are the value of the property, the amount of insurance coverage, and the amount of damage.

```
10 INPUT"ENTER VALUE OF PROPERTY";V
20 INPUT"ENTER AMOUNT OF INSURANCE COVERAGE";C
30 INPUT"ENTER DAMAGE AMOUNT";D
40 X=C/V
50 IF X>=.80 THEN 80
60 D=D*X
70 GOTO 100
```

```
80 Y=D/V
90 IF Y>=X THEN D=V*X
100 D=INT(D*100+.5)/100
110 PRINT"AMOUNT INSURANCE COMPANY PAYS: $";D
```

SAMPLE PROBLEM

If you insured your home for $72,500 and its true value is $103,250, how much would your co-insurance policy pay if your home sustained damage estimated at $32,000?

```
ENTER VALUE OF PROPERTY ? 103250
ENTER AMOUNT OF INSURANCE COVERAGE ? 72500
ENTER DAMAGE AMOUNT ? 32000
AMOUNT INSURANCE COMPANY PAYS: $ 22469.73
```

COST OF AUTOMOBILE INSURANCE

The majority of automobile insurance companies use charts and tables to compute the premium for an insurance policy. There are rates according to age, sex, driving record, and other statistical data. The method shown here for computing the premium is based on an equation. This is not a standard formula, but just an example of how this can be done. Any insurance company could easily structure an equation for the possible variables that the company considers when quoting rates. This program could be altered for the desired situation by changing the factors .2, .3, and .1 in lines 70, 80, and 90, respectively. Also, other lines could be added for additional insurance considerations.

The following program computes the amount of an automobile insurance premium. The inputs to the program are the amount of the insurance coverage and statistical information about the person being covered.

```
10 INPUT"ENTER MALE(M) OR FEMALE(F)";S$
20 INPUT"ENTER AGE";A
30 INPUT"ENTER DRIVING RECORD GOOD(G) OR BAD(B)";R$
40 INPUT"ENTER AMOUNT OF INSURANCE COVERAGE";C
50 X=C/100
60 Z=X
70 IF S$="M" THEN Z=Z+X*.2
80 IF A<21 THEN Z=Z+X*.3
90 IF R$="B" THEN Z=Z+X*.1
```

```
100 Z=INT(Z*100+.5)/100
110 PRINT"PREMIUM AMOUNT: $";Z
```

SAMPLE PROBLEM

If an 18-year-old female with a bad driving record needed $50,000 worth of automobile insurance, what would be the premium?

```
ENTER MALE(M) OR FEMALE(F) ? F
ENTER AGE ? 18
ENTER DRIVING RECORD GOOD(G) OR BAD(B) ? B
ENTER AMOUNT OF INSURANCE COVERAGE ? 50000
PREMIUM AMOUNT: $700
```

FUTURE VALUE OF LIFE INSURANCE PAYMENTS

Many people buy life insurance to provide financial protection for their families in the event of death. There are numerous kinds of life insurance, but all require the payment of some type of periodic payment.

It is interesting to look at a life insurance policy as an annuity. The program shown here determines the future value of the life insurance premiums if they were put into a savings account type of annuity.

The following program computes the future value of premium payments, given the amount of the monthly premium payment, the interest rate, and the number of months that the premium is paid.

```
10 INPUT"ENTER AMOUNT OF INSURANCE PREMIUM";PM
20 INPUT"ENTER INTEREST RATE AVAILABLE";K
30 INPUT"ENTER NUMBER OF MONTHS PREMIUM PAID";N
40 K=K/1200
50 FV=PM*(((1+K)↑N-1)/K)
60 FV=INT(FV*100+.5)/100
70 PRINT"FUTURE VALUE OF PREMIUM PAYMENTS: $";FV
```

SAMPLE PROBLEM

If you have a life insurance policy that requires you to pay a monthly premium of $25.35, what would be the future value of these payments after 20 years (240 months) if the current interest rate is 9%?

```
ENTER AMOUNT OF INSURANCE PREMIUM ? 25.35
ENTER INTEREST RATE AVAILABLE ? 9
ENTER NUMBER OF MONTHS PREMIUM PAID ? 240
FUTURE VALUE OF PREMIUM PAYMENTS: $ 16930.93
```

CHAPTER

14

Metric Conversion

The metric system is the standard for measurements of length, weight, and volume in most worldwide markets. The United States still has not adopted these standard measurements for everyday activities, but since most businesses have to relate their product specifications to metric standards, it's often necessary to convert the American units of measure to the metric system.

The standard metric measurements for length, weight, and volume are the meter, gram, and liter, respectively. Factors of ten of each of these measurements are denoted by a prefix placed before the unit measurement, such as millimeter and kilometer.

When the conversions from one measurement system to another are done in this chapter, the measure may not be exactly correct, due to round-off error and the number of decimal places used in the conversion factor.

ENGLISH LENGTH UNITS TO METRIC

The standard unit of length in the metric system is the meter. This corresponds to various length measurements in the English system, such as inch, foot, yard, and mile.

The following program computes the equivalent number of meters, given an inch, foot, yard, or mile measurement. The user is prompted for the English measurement desired.

```
10 PRINT"SPECIFY LENGTH MEASUREMENT DESIRED"
20 INPUT"1-INCH, 2-FEET, 3-YARD, 4-MILE";X
30 INPUT"ENTER NUMBER OF UNITS";U
40 ON X GOTO 50,70,90,110
```

```
50 Z=.025*U
60 GOTO 120
70 Z=.31*U
80 GOTO 120
90 Z=.91*U
100 GOTO 120
110 Z=1600*U
115 PRINT"EQUIVALENT NUMBER OF KILOMETERS: ";Z/1000
120 PRINT"EQUIVALENT NUMBER OF METERS: ";Z
```

SAMPLE PROBLEM

You are planning a trip and plan to travel 620 miles. How many meters would this be?

```
SPECIFY LENGTH MEASUREMENT DESIRED
1-INCH, 2-FEET, 3-YARD, 4-MILE ? 4
ENTER NUMBER OF UNITS ? 620
EQUIVALENT NUMBER OF KILOMETERS: 992
EQUIVALENT NUMBER OF METERS: 992000
```

ENGLISH WEIGHT UNITS TO METRIC

The standard unit of weight in the metric system is the gram. This corresponds to various methods of weight measurements in the English system, such as the ounce and the pound.

The following program computes the equivalent number of grams, given an ounce or pound measurement. The user is prompted for the English measurement desired.

```
10 PRINT"SPECIFY WEIGHT MEASUREMENT DESIRED"
20 INPUT"1-OUNCE, 2-POUND";X
30 INPUT"ENTER NUMBER OF UNITS";U
40 ON X GOTO 50,70
50 Z=28*U
60 GOTO 80
70 Z=454*U
80 PRINT"EQUIVALENT NUMBER OF GRAMS: ";Z
```

SAMPLE PROBLEM

If you purchased 5 pounds of sugar, how many grams would this be?

```
SPECIFY WEIGHT MEASUREMENT DESIRED
1-OUNCE, 2-POUND ? 2
ENTER NUMBER OF UNITS ? 5
EQUIVALENT NUMBER OF GRAMS: 2270
```

ENGLISH VOLUME UNITS TO METRIC

The standard unit of volume in the metric system is the liter. This corresponds to various methods of volume measurements in the English system, such as the pint, quart, and gallon.

The following program computes the equivalent number of liters, given a pint, quart, or gallon measurement. The user is prompted for the English measurement desired.

```
10 PRINT"SPECIFY VOLUME MEASUREMENT DESIRED"
20 INPUT"1-PINT, 2-QUART, 3-GALLON";X
30 INPUT"ENTER NUMBER OF UNITS";U
40 ON X GOTO 50,70,90
50 Z=.47*U
60 GOTO 100
70 Z=.95*U
80 GOTO 100
90 Z=3.8*U
100 PRINT"EQUIVALENT NUMBER OF LITERS: ";Z
```

SAMPLE PROBLEM

If you wanted to purchase 10 gallons of gasoline, but it was sold on a liter basis, how many liters would be needed?

```
SPECIFY VOLUME MEASUREMENT DESIRED
1-PINT, 2-QUART, 3-GALLON ? 3
ENTER NUMBER OF UNITS ? 10
EQUIVALENT NUMBER OF LITERS: 38
```

METRIC LENGTH UNITS TO ENGLISH

The standard metric measurement for length is the meter. A specified number of meters can be broken down into the equivalent number of English length units.

The following program computes the equivalent number of inches, feet, yards, and miles, given a specified number of meters.

```
10 INPUT"ENTER NUMBER OF METERS";M
20 X=M*39.37
30 PRINT"EQUIVALENT # OF INCHES: ";X
40 X=M*3.28
50 PRINT"EQUIVALENT # OF FEET: ";X
60 X=M*1.1
70 PRINT"EQUIVALENT # OF YARDS: ";X
80 X=M*.000620
90 PRINT"EQUIVALENT # OF MILES: ";X
```

SAMPLE PROBLEM

If you are running in a 10,000 meter race, what are the equivalent English measurements for this distance?

```
ENTER NUMBER OF METERS ? 10000
EQUIVALENT # OF INCHES: 393700
EQUIVALENT # OF FEET: 32800
EQUIVALENT # OF YARDS: 11000
EQUIVALENT # OF MILES: 6.2
```

METRIC WEIGHT UNITS TO ENGLISH

The standard metric measurement for weight is the gram. A specified number of grams can be broken down into the equivalent number of English weight units.

The following program computes the equivalent number of ounces and pounds, given a specific number of grams.

```
10 INPUT"ENTER NUMBER OF GRAMS";G
20 X=G*.035
30 PRINT"EQUIVALENT # OF OUNCES: ";X
40 X=G*.0022
50 PRINT"EQUIVALENT # OF POUNDS: ";X
```

SAMPLE PROBLEM

If you were preparing a dessert where the recipe called for 2,000 grams of bananas, what would be the equivalent English measurements for this weight?

```
ENTER NUMBER OF GRAMS ? 2000
EQUIVALENT # OF OUNCES: 70
EQUIVALENT # OF POUNDS: 4.4
```

METRIC VOLUME UNITS TO ENGLISH

The standard metric measurement for volume is the liter. A specified number of liters can be broken down into the equivalent number of English volume units.

The following program computes the equivalent number of pints, quarts, and gallons, given a specified number of liters.

```
10 INPUT"ENTER NUMBER OF LITERS";L
20 X=L*2.1
30 PRINT"EQUIVALENT # OF PINTS: ";X
40 X=L*1.06
50 PRINT"EQUIVALENT # OF QUARTS: ";X
60 X=L*.26
70 PRINT"EQUIVALENT # OF GALLONS: ";X
```

SAMPLE PROBLEM

If you purchase a car with a 50-liter gas tank, what are the equivalent English measurements for this volume?

```
ENTER NUMBER OF LITERS ? 50
EQUIVALENT # OF PINTS: 105
EQUIVALENT # OF QUARTS: 53
EQUIVALENT # OF GALLONS: 13
```

COMPARING METER TO OTHER METRIC LENGTHS

The standard unit of length in the metric system is the meter. The other metric length units are all multiples of 10 relative to this standard unit. The decimal point is the only thing that will change when doing a conversion from one unit measurement to another.

The following program computes the other metric length measurements, given a specified number of meters.

```
10 INPUT"ENTER NUMBER OF METERS";M
20 X=M*10
```

```
30 PRINT"EQUIVALENT DECIMETERS: ";X
40 X=M*100
50 PRINT"EQUIVALENT CENTIMETERS: ";X
60 X=M*1000
70 PRINT"EQUIVALENT MILLIMETERS: ";X
80 X=M*.1
90 PRINT"EQUIVALENT DEKAMETERS: ";X
100 X=M*.01
110 PRINT"EQUIVALENT HECTOMETERS: ";X
120 X=M*.001
130 PRINT"EQUIVALENT KILOMETERS: ";X
```

SAMPLE PROBLEM

What is the equivalent of 2,000 meters in other metric length units?

```
ENTER NUMBER OF METERS ? 2000
EQUIVALENT DECIMETERS: 20000
EQUIVALENT CENTIMETERS: 200000
EQUIVALENT MILLIMETERS: 2000000
EQUIVALENT DEKAMETERS: 200
EQUIVALENT HECTOMETERS: 20
EQUIVALENT KILOMETERS: 2
```

COMPARING LITER TO OTHER METRIC VOLUMES

The standard unit of volume in the metric system is the liter. The other metric volume units are all multiples of 10 relative to this standard unit. The decimal point is the only thing that will change when doing a conversion from one unit measurement to another.

The following program computes the other metric volume measurements, given a specified number of liters.

```
10 INPUT"ENTER NUMBER OF LITERS";L
20 X=L*10
30 PRINT"EQUIVALENT DECILITERS: ";X
40 X=L*100
50 PRINT"EQUIVALENT CENTILITERS: ";X
60 X=L*1000
70 PRINT"EQUIVALENT MILLILITERS: ";X
```

```
80 X=L*.1
90 PRINT"EQUIVALENT DEKALITERS: ";X
100 X=L*.01
110 PRINT"EQUIVALENT HECTOLITERS: ";X
120 X=L*.001
130 PRINT"EQUIVALENT KILOLITERS: ";X
```

SAMPLE PROBLEM

What is the equivalent of 149 liters in other metric volume units?

```
ENTER NUMBER OF LITERS ? 149
EQUIVALENT DECILITERS: 1490
EQUIVALENT CENTILITERS: 14900
EQUIVALENT MILLILITERS: 149000
EQUIVALENT DEKALITERS: 14.9
EQUIVALENT HECTOLITERS: 1.49
EQUIVALENT KILOLITERS: .149
```

COMPARING GRAM TO OTHER METRIC WEIGHTS

The standard unit of weight in the metric system is the gram. The other metric weight units are all multiples of 10 relative to this standard unit. The decimal point is the only thing that will change when doing a conversion from one unit measurement to another.

The following program computes the other metric weight measurements, given a specified number of grams.

```
10 INPUT"ENTER NUMBER OF GRAMS";G
20 X=G*10
30 PRINT"EQUIVALENT DECIGRAMS: ";X
40 X=G*100
50 PRINT"EQUIVALENT CENTIGRAMS: ";X
60 X=G*1000
70 PRINT"EQUIVALENT MILLIGRAMS: ";X
80 X=G*.1
90 PRINT"EQUIVALENT DEKAGRAMS: ";X
100 X=G*.01
110 PRINT"EQUIVALENT HECTOGRAMS: ";X
120 X=G*.001
130 PRINT"EQUIVALENT KILOGRAMS: ";X
```

Sample Problem

What is the equivalent of 57 grams in other metric weight units?

```
ENTER NUMBER OF GRAMS ? 57
EQUIVALENT DECIGRAMS: 570
EQUIVALENT CENTIGRAMS: 5700
EQUIVALENT MILLIGRAMS: 57000
EQUIVALENT DEKAGRAMS: 5.7
EQUIVALENT HECTOGRAMS: .57
EQUIVALENT KILOGRAMS: .057
```

Customizing an Application

Numerous small programs can be combined into a customized application to fit your personal needs. This can be done by selecting the specific applications you desire and then setting up a menu-driven option list to branch to the specific applications.

An example of this customizing method is demonstrated using four programs from chapter 8, "Time Value of Money." These programs are:

1. Future Value of a Single Deposit
2. Present Value of a Future Receipt
3. Future Value of an Ordinary Annuity
4. Present Value of an Ordinary Annuity

The menu is initially displayed, followed by a prompt for the function desired. This portion of the program is shown here.

```
10 PRINT
20 PRINT"CUSTOMIZED APPLICATIONS:"
30 PRINT"------------------------"
40 PRINT"ENTER-1 FV OF SINGLE DEPOSIT"
50 PRINT"ENTER-2 PV OF FUTURE RECEIPT"
60 PRINT"ENTER-3 FV OF ORDINARY ANNUITY"
70 PRINT"ENTER-4 PV OF ORDINARY ANNUITY"
80 PRINT"ENTER-5 TO EXIT"
90 INPUT IN
100 ON IN GOTO 1000,2000,3000,4000,5000
110 GOTO 10
```

This partial program listing is followed by each of the individual listings for each topic as shown in chapter 8. After each function is complete, control is returned to the start of the program to redisplay the menu. The remainder of the program follows.

```
999 '*** THE FOLLOWING PGM IS FROM CHP 8 SECTION 1 ***
1000 INPUT"ENTER INITIAL VALUE";A
1010 INPUT"ENTER INTEREST RATE";K
1020 INPUT"ENTER # OF COMPOUND PERIODS/YEAR";M
1030 INPUT"ENTER NUMBER OF YEARS INVESTED";N
1040 K=K/100
1050 FV=A*(1+K/M)↑(N*M)
1060 FV=INT(FV*100+.5)/100
1070 PRINT"FUTURE VALUE: $";FV
1080 GOTO 10
1999 '*** THE FOLLOWING PGM IS FROM CHP 8 SECTION 2 ***
2000 INPUT"ENTER FUTURE RECEIPT";FV
2010 INPUT"ENTER INTEREST RATE";K
2020 INPUT"ENTER NUMBER OF YEARS";N
2030 K=K/100
2040 PV=FV*(1+K)↑-N
2050 PV=INT(PV*100+.5)/100
2060 PRINT"PRESENT VALUE: $";PV
2070 GOTO 10
2999 '*** THE FOLLOWING PGM IS FROM CHP 8 SECTION 3 ***
3000 INPUT"ENTER PAYMENT AMOUNT";PM
3010 INPUT"ENTER INTEREST RATE";K
3020 INPUT"ENTER NUMBER OF MONTHS";N
3030 K=K/1200
3040 FV=PM*(((1+K)↑N-1)/K)
3050 FV=INT(FV*100+.5)/100
3060 PRINT"FUTURE VALUE: $";FV
3070 GOTO 10
3999 '*** THE FOLLOWING PGM IS FROM CHP 8 SECTION 4 ***
4000 INPUT"ENTER PAYMENT AMOUNT";PM
4010 INPUT"ENTER INTEREST RATE";K
4020 INPUT"ENTER NUMBER OF MONTHS";N
4030 K=K/1200
4040 PV=PM*((1-(1+K)↑-N)/K)
4050 PV=INT(PV*100+.5)/100
4060 PRINT"PRESENT VALUE: $";PV
4070 GOTO 10
5000 END
```

As can be seen with this example, each of the four programs from chapter 8 was duplicated in this program. Depending on the initial selection by the user of the program, control is given to one of the four applications. Once the short application is completed, control is once again returned to the menu for another selection.

6457-8

Taking Care of Business with Your Commodore 64®

David P. Dautenhahn

If taking care of home financial matters has gotten to be one big pain in the neck, now you can make your life easier by *Taking Care of Business with Your Commodore 64.*

Using the business and financial programs provided in this book, you'll be able to see to all your financial needs quickly, easily, and accurately. Each of the programs is carefully documented with a short explanation of what the computer will do and a BASIC listing. Then, to further aid your understanding, a real-life scenario is given along with a sample run. Instruction on how to combine two or more of these applications is also provided.

Programs include: • Interest • Depreciation • Retailing • Real Estate • Loan Analysis • Savings • Lease Analysis • Time Value of Money • Stocks and Bonds Analysis • Sinking Fund Analysis • Forecasting Inventory Needs • Payroll • Insurance • Metric Conversion.

So, whether you want to catch up on work from the office, or you're just interested in managing household finances, *Taking Care of Business with Your Commodore 64* is a great place to start.

Also of interest…

Basic Commodore 64™ BASIC
JAMES S. COAN

The same simple, direct approach that made the first and all other versions instant best sellers. Readers learn how to build complex application programs from short, simple ones. Special features and advanced programming techniques are clearly explained. More than 80 programs in all! #6456-X, paper, 272 pages.

ALLEN COUNTY PUBLIC LIBRARY
3 1833 04371 1156

HAYDEN BOOK
a division of Hayden Publishing Company, Inc.
Hasbrouck Heights, New Jersey
ISBN 0-8104-6457-8